


DICK SMITH

VZ200

**Personal
Colour Computer**

**First
Book of Programs**

D.M WAS HERE 85.



B-7202

DICK SMITH'S FIRST BOOK OF PROGRAMS FOR THE VZ200



FOREWORD FROM DICK SMITH

Incredible! I just can't seem to keep up with the way these computers continue to get not only smaller, but also more intelligent. With the release of our amazing new VZ-200¹³⁰⁰ computer, we have continued our commitment to making available what we believe are some of the most exciting advances in computer technology to be seen in years.

Who would have imagined just a few short years ago that we would be able to produce a machine like the VZ-200 with features such as colour, graphics, sound and BASIC from the well respected American firm, Microsoft. Add to that outputs for either T.V. or monitor, provision for cassette storage, memory expansion and a printer interface. Take all this, pack it into a small, neat package and sell it for an unbelievably low price of just \$199.00. Whoever would have believed it!

In brief, this then is the amazing VZ-200 and as far as value for money is concerned, I believe that it will be very hard to beat.

The VZ-200 is primarily aimed at the person who wants to 'get into' computing and learn 'what it's all about' without having the normal hassles associated with larger, more expensive machines.

Regardless of whether you are young or classed as an 'oldie' (by others that is), you will benefit from the many facets of learning that can be enjoyed using your VZ-200. You will quickly appreciate the diversity of capability and you will find that as you learn one facet of the VZ-200, it will lead you into another, each time increasing your own ability. You may for example be interested in producing colourful graphics designs, simple music or like to simply 'muck about' with programming. It doesn't really matter, so long as you have fun doing it.

To enable us therefore to give you a really good introduction, and to assist by taking some of the mystery out of computer programming, we have produced this book of programs to help show you 'how it's done'. Some are entertaining and some are educational. We hope you enjoy the programs presented herein and while you are learning about your new computer, remember to have fun!

Here's little
Dick Smith

ALL on tape 1
Sides

- First/second book of Progs

Q some ON B Match Hangman
V=done it not ALL on same tape

* on tape no 1

TABLE OF CONTENTS

First Book ON

First tape →
Print out - X

TITLE	PAGE
* CRAZY STORY.....	1
* VZ-200 JUMBLE CODE.....	5
* WORD RACE.....	9
* WHAT'S MY NUMBER.....	14
* WHAT'S MY LETTER.....	16
* VZ-200 MATHS QUIZ.....	18
* VZ-200 MUSIC BOX.....	22
* VZ-200 CLOCK.....	25
* WHAT DAY OF THE WEEK IS IT.....	28
* CRYSTAL SPECTACULAR.....	31
* WRITE A STORY.....	33
* VZ-200 HANGMAN.....	38
* MOVING TARGETS.....	45
* VZ-200 MASTERMIND.....	50
* VZ-200 CUP.....	54
VIDEO DISPLAY WORKSHEETS.....	59
NOTE PAGE.....	60

Second book of Progs At 200

Poke 30862, 20. Pok = 30863, 52. x = VSR (*)

dark light correct colour & position
correct colour only

DICK SMITH'S FIRST BOOK OF PROGRAMS FOR THE VZ200

Programs written by Wayne G. Richmond
Compiled by Cary M. Laue'

Cartoons by Andrew Barrance

National Library Card of Australia No.
and ISBN 0 949772 17 8

COPYRIGHT (C) 1983, DICK SMITH MANAGEMENT

The material contained in this book is protected by copyright. It may not be legally reproduced - stored in a retrieval system, transmitted or copied by any means - - whether electrical, magnetic, photographic or any other technology - - except for private use by the owner - - without the written permission of the publisher. All such rights are reserved by Dick Smith Management Pty Ltd, Sydney, Australia.

Crazy Story



As the first program in our book for the VZ-200, we decided it would be a good idea to get you laughing. We certainly hope that it does!

This program jumbles up parts of a series of 10 stories with crazy results. The stories are then displayed on your screen and if you have a printer attached to your computer, you can choose to print the story out on paper and even produce your own book!

FOR THE PROGRAMMER

As this book has been written with the intention of helping you to learn more about the programming side of your VZ-200, we will present in this segment for each program, information relating to 'how it's done' in the program.

It is also important to note that where the little pointer  is used in the margin of a program line, it means that the information needed to determine how we achieved the particular graphics code used in that program line can be found at the end of each program. This mainly relates to the use of inverse or graphics characters fed in directly from the keyboard.

In this first program, note the routine used at lines 20040-20070 which creates the MARQUEE effect.

```
10 REM * CRAZY STORY *
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (MAY 1983)
35 CLS: COLOR ,0
40 DIM A1$(10), A2$(10), A3$(10), A4$(10), A5$(10)
50 FOR N = 0 TO 10:READ A1$(N):NEXT
60 FOR N = 0 TO 10:READ A2$(N):NEXT
80 FOR N = 0 TO 10:READ A3$(N):NEXT
90 FOR N = 0 TO 10:READ A4$(N):NEXT
100 FOR N = 0 TO 10:READ A5$(N):NEXT
110 IF PEEK(28416) < 200 THEN PT$ = "NO PRINTER" ELSE PT$ = ""
120 GOSUB 20000 ' (INTRODUCTION)
1000 REM MAIN ROUTINE
1010 GOSUB 40000 ' (CLEAR SCREEN)
1020 A1 = RND(11) - 1
1030 A2 = RND(11) - 1
1040 A3 = RND(11) - 1
1050 A4 = RND(11) - 1
1060 A5 = RND(11) - 1
1100 PRINT@162,A1$(A1);
1110 PRINT@194,A2$(A2);
```

```


1120 IF LEN(A3$(A3)+A4$(A4)) > 28 THEN 1160
1130 PRINT@226,A3$(A3);" ";A4$(A4);
1140 PRINT@258,A5$(A5);".";
1150 GOTO 1200
1160 PRINT@226,A3$(A3);
1170 PRINT@258,A4$(A4);
1180 PRINT@290,A5$(A5);".";
1200 PRINT@417," ";
1210 K$ = INKEY$
1220 I$ = INKEY$:IF I$ = "" THEN 1220
1230 IF I$ = "P" THEN 1500
1240 IF I$ <> " " THEN 1220
1250 SOUND 30,1
1260 GOTO 1000
1500 REM PRINT STORY ON PRINTER
1510 T = 16 '(SET TAB TO 16)
1520 IF PT$ = "NO PRINTER" THEN 1220
1530 PRINT@417," ";
1540 SOUND 1,1
1545 IF TI = 99 THEN 1600 '(SKIP TITLE)
1548 LPRINT
1550 LPRINTTAB(26) "*****"
1560 LPRINTTAB(26) "* C R A Z Y S T O R I E S *"
1570 LPRINTTAB(26) "*****":LPRINT
1575 LPRINT:LPRINT
1580 TI = 99' (SET TITLE FLAG)
1600 LPRINT TAB(T)A1$(A1);" ";A2$(A2);" ";A3$(A3)
1610 LPRINT
1620 LPRINT TAB(T) A4$(A4);" ";A5$(A5);"."
1630 LPRINT:LPRINT:LPRINT:LPRINT
1660 K$ = INKEY$
1670 GOTO 1200
10000 DATA ALMOST ALL OF THESE,ALL OF MY,MORE OF HIS,SOME OF THE
10010 DATA PLENTY OF,MOST OF YOUR,SIX OF MY,NEARLY ALL
10020 DATA MANY OF THOSE,SEVERAL OF OUR,A FEW OF HER
10100 DATA STINKING MAGGOTY CHEESES,POOR BEDRAGGLED DUCKS
10110 DATA ROTTEN YELLOW TURNIPS,LIVELY YOUNG TEACHERS
10120 DATA CHARMING OLD GENTLEMEN,BEST-BEHAVED CHILDREN
10130 DATA MAN-EATING TIGERS,LITTLE BOY'S TROUSERS
10140 DATA SCENTED PINK ROSES,OLD-FASHIONED DUCHESSSES
10150 DATA LONG-HAIRED SHEEPDOGS
10160 DATA OUGHT TO BE,SHOULD BE,COULD BE,MAY BE,LIKE TO BE
10170 DATA NICHT BE,MUST BE,NEED TO BE,HAVE TO BE,ARE TO BE
10180 DATA CAN BE
10200 DATA REMOVED AT ONCE & DUMPED,CAREFULLY SHAMPOOED
10210 DATA SLICED UP AND STEWED,FOUND PAINTING PICTURES
10220 DATA LOVINGLY CARED FOR,SENT ON A MYSTERY TOUR
10230 DATA CAUGHT AND EXHIBITED,TURNED UPSIDE DOWN & SHAKEN
10240 DATA SPRAYED AGAINST FRUITFLY,DRESSED IN SILKS
10250 DATA COMBED & TREATED FOR FLEAS
10300 DATA IN THE NEAREST DUSTBIN,TO REMOVE THE GREASE
10400 DATA BEFORE BEING FED TO THE PIGS,ALL OVER THE SCHOOL
10410 DATA BY THE ATTRACTIVE NURSES,AS A SPECIAL TREAT

```

```

10420 DATA IN A CAGE AT THE LOCAL ZOO
10430 DATA BEFORE BEING PUT IN THE WASH, IN THE MIDDLE OF SUMMER
10440 DATA LIKE THE QUEEN OF SHEBA, WITHOUT MAKING A FUSS
11000 DATA SELECT THE SMALLEST ONES, SPEAK TO THEM POLITELY
11010 DATA HANG THEM ON THE LINE, BOIL UNTIL TENDER
11020 DATA STALK THEM SILENTLY, PANDAGE UP WITH CREEN HEPES
11030 DATA FIND WHERE THEY GATHER, PUT THEM INTO HARNESS
11040 DATA COLLECT SOME FROM THE WOODS, WASH THEM IN COLD WATER
11050 DATA SPREAD THEM ON A BAKING DISH
11060 DATA CHOOSE SOME OF DIFFERENT COLOURS
11070 DATA PUT THEM INTO PICKLE JARS, FORGET YOUR SENSE OF HUMOUR
11080 DATA BEAT THEM VIGOROUSLY, DRAIN THEM VERY CAREFULLY
11090 DATA WAIT TILL THEY SETTLE, LEAVE FOR A FORTNIGHT
11100 DATA DUST THEM WITH FLEA POWDER, PLANT THEM IN A SHADY SPOT
11110 DATA TROT ROUND IN CIRCLES WITH THEM, BLOW THEM ALL UP
11120 DATA REMOVE ALL THE LEAVES & ROOTS, PAINT THEM WITH EGG YOLK
12000 DATA COVER THEM WITH VINEGAR, TELL SOME OLD JOKES
12010 DATA SWEEP WITH A STIFF BRUSH, SPRINKLE THEM WITH CHEESE
12020 DATA SWAT THEM WITH A NEWSPAPER, HOPE FOR THE BEST
12030 DATA STAMP ON THOSE STILL ACTIVE, WATER THEM DAILY
12040 DATA GIVE THEM SUGAR AT INTERVALS, BAKE IN A HOT OVEN
12050 DATA SERVE THEM ON A LETTUCE LEAF
12060 DATA STRING THEM AROUND THE ROOM
20000 REM INTRODUCTION
20010 GOSUB 30000' (DRAW FRAME)
20020 COLOR 7
20030 PRINT@73, " " ;
20040 T$ = "(BY WAYNE G. RICHMOND)"
20050 FOR N = 1 TO LEN(T$)
20060   PRINT@133, RIGHT$(T$, N);
20070 NEXT
20100 PRINT@226, "THIS PROGRAM RANDOMLY";
20110 PRINT@290, "PRODUCES NONSENSE STORIES!";
20120 PRINT@419, " " ;
20130 K$ = INKEY$
20140 I$ = INKEY$: IF I$ <> " " THEN 20140
20150 SOUND 30, 1
20160 RETURN
20210 REM
30000 REM FRAME
30010 CLS
30020 COLOR 8
30030 PRINT@0, " " ;
30040 PRINT@480, " " ;
30050 FOR N = 32 TO 448 STEP 32
30060   PRINT@N, " " ;
30070   PRINT@N+31, " " ;
30080 NEXT
30090 PRINT@31, " " ;
30100 POKE 29183, 184
30110 RETURN

```



```

40000 REM CLEAR SCREEN
40010 FOR N = 130 TO 322 STEP 32
40020 PRINT@N,"
40030 NEXT
40040 RETURN

```

EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

As noted at the beginning of this program, we will, at the end of each program in this book, detail a list of the program lines that use graphics codes and show you how they have been achieved.

Being the first program, we will explain in more detail how each individual code is created, however, in all other programs, only abbreviated key codes will be used.

Initially, a line of graphics code will (usually) be either INVERSE printing or a keyboard generated set of graphics codes. To achieve the INVERSE code such as that used in line 1200, press CTRL (CONTROL) first and while holding CTRL, press INVERSE. You can then type in the letters of the required heading etc. To 'switch off' the inverse mode, simply repeat the procedure of CTRL and INVERSE as before. Note that in the text that follows in which we describe the keys used, the symbol will for example, be denoted as CTRL + INV + TEXT:

RETURN

To achieve graphics symbols or lines of graphics such as at line 30030, simply press the SHIFT key and while holding the SHIFT key, press the required graphics 'block' key (O to I and A to J). Note that in the text that follows in which we describe the keys used, the symbol for a specific graphic code will for example, be denoted as SHIFT + B.


Lastly, note that all graphics codes, inverse code and text, will always be keyed in, within inverted commas.

Line 1200	CTRL + INV + TEXT: (SPACES) = NEW STORY P:PRINT
Line 1530	CTRL + INV + TEXT: (PRINTING STORY)
Line 20030	CTRL + INV + TEXT: (CRAZY STORY)
Line 20120	CTRL + INV + TEXT: PRESS ((SPACE)) TO BEGIN
Line 30030	SHIFT + A, then SHIFT + Y's, then SHIFT + S
Line 30040	SHIFT + D, then SHIFT + T's, then SHIFT + F
Line 30060	SHIFT + I
Line 30070	SHIFT + U
Line 30090	SHIFT + S


```

1200 IF I$ = "3" THEN LW = 0:WP = 173
1210 IF I$ = "4" THEN LW = 20:WP = 172
1220 IF I$ = "5" THEN LW = 40:WP = 172
1230 IF I$ = "6" THEN LW = 60:WP = 171
1240 IF I$ = "7" THEN LW = 80:WP = 170
1250 IF I$ = "8" THEN LW = 100:WP = 169
1300 REM SELECT WORD
1310 W$ = W$(RND(20)+LW)
1320 L = LEN(W$)
1330 GOSUB 10000 '(CLEAR SCREEN)
2000 REM JUMBLE LETTERS
2005 CW$ = ""
2007 FORN=1TOL:R(N)=0:NEXT
2010 FOR N = 1 TO L
2020   R = RND(L)
2030   FOR X = 1 TO N
2040     IF R = R(X) THEN 2020
2050   NEXT X
2060   R(N) = R
2070   CW$ = CW$ + MID$(W$,R,1)
2080 NEXT N
2100 REM CHECK THAT WORD IS JUMBLD
2110 IF CW$ = W$ THEN 2000
3000 REM PRINT WORD ON SCREEN
3010 PRINT@WP,;
3020 FOR N = 1 TO L
3030   PRINTMID$(W$,R(N),1);" ";
3040 NEXT
3050 PRINT@364,"          ";
3060 PRINT@418,"          ";
3070 PRINT@354,"GUESS";K;
3080 INPUT G$:G$ = LEFT$(G$,L)
3090 IF G$ = "" THEN 3070
3100 IF G$ = W$ THEN 5000 '(CORRECT ROUTINE)
3110 K = K + 1
3120 PRINT@373," ***** ";
3130 SOUND 1,6
3140 IF K = 7 THEN 6000 '(LOSE ROUTINE)
3150 IF K = 7 THEN 6000 '(LOSE ROUTINE)
3160 IF K = 7 THEN 6000 '(LOSE ROUTINE)
3170 IF K < 6 THEN PRINT@418," ***** ";GOTO 2000
3180 PRINT@418," ***** ";GOTO 2000
5000 REM CORRECT ROUTINE
5010 PRINT@372," ***** ";
5020 FOR N = 1 TO 3
5030   SOUND 28,1
5040   SOUND 25,1
5050 NEXT
5060 SOUND 28,6
5070 GOTO 7000

```



EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

Line 1030 CTRL + INV + TEXT: VZ-200 JUMBLE CODE
Line 3120 CTRL + INV + TEXT: WRONG !
Line 3170 CTRL + INV + TEXT: TRY AGAIN !
Line 3180 CTRL + INV + TEXT: LAST CHANCE !
Line 5010 CTRL + INV + TEXT: CORRECT !
Line 6010 CTRL + INV + TEXT: YOU BOMBED !!
Line 7050 CTRL + INV + TEXT: PRESS (SPACE) FOR NEW GAME
Line 20020 SHIFT + A, then SHIFT + Y's, then SHIFT + S
Line 20030 SHIFT + D, then SHIFT + T's, then SHIFT + F
Line 20050 SHIFT + I
Line 20060 SHIFT + U
Line 20080 SHIFT + S

Word Race


This program is an exciting race against the clock to sort a list of six words into alphabetical order. The faster you do it, the higher your score. The computer will keep track of the name and score of the player with the highest score. Good luck!

FOR THE PROGRAMMER

In order to determine which keys are being pressed we use the PEEK command to examine location 28416 (line 103). The value stored in this location will determine which key has been pressed.

If you have the 16K memory expansion module fitted, it would be a good idea to add more words to the program. To do this simply add extra DATA lines and alter (by increasing) the value of NW in line 90. You may also have to increase the amount of space CLEARED for strings in line 60.

```
10 REM * WORD RACE *
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (APRIL 1983)
50 CLS
60 CLEAR 200
90 NW = 56 'NUMBER OF WORDS
100 DIM W$(NW)
110 CD = 6000 '(SET COUNTDOWN)
120 CP = 80 '(SET CURSOR POSITION)
130 COLOR 4 '(SET RED)
140 DN = 2000 '(SET DELAY TIMER)
200 FOR N = 1 TO NW
210 READ W$(N)
220 NEXT
300 GOSUB 31000 '(INSTRUCTIONS)
310 GOSUB 32000 '(INPUT NAME)
320 GOSUB 35000 '(CLEAR SCREEN)
330 GOSUB 20000 '(SELECT LIST)
340 GOSUB 2000 '(PRINT LIST)
1000 REM PEEK KEYBOARD
1002 IF TS = 0 THEN 1015
1005 PRINT@151,;
1010 PRINTUSING" ( #,###) ";TS
1015 PO=66:G=1:W=1
1020 PRINT@PO, "###";
1030 KB=PEEK(28416)
1035 GOSUB 6000
```



```


1040 IF KB = 251 THEN GOSUB 5000
1050 IF KB <> 239 AND KB <> 253 THEN 1030
1055 SOUND RND(32)-1,1
1060 PRINT@PO," ";
1070 IF KB = 239 THEN PO = PO + 64:W = W + 1:GOTO 1100
1080 IF KB = 253 THEN PO = PO - 64:W = W - 1:GOTO 1200
1100 IF PO > 386 THEN PO = 66:W = 1
1110 GOTO 1020
1200 IF PO < 66 THEN PO = 386:W = 6
1210 GOTO 1020
2000 REM PRINT WORDS
2005 P=68
2020 FOR N = 1 TO 6
2030 PRINT@P,W$(R(N))
2040 P=P+64
2050 NEXT
2060 RETURN
5000 REM CHECK ANSWER
5010 IF W$(R(W)) = A$(G) THEN 7000 ELSE 8000
6000 REM PRINT COUNTDOWN
6010 CD=CD-5
6030 PRINT@88,;
6035 PRINTUSING"#,###";CD
6040 IF CD <= 0 THEN 9000
6050 RETURN
7000 REM PRINT CORRECT WORD
7020 PRINT@CP,W$(R(W));
7030 PRINT@PO+2," ";
7040 FOR N = 1 TO 1
7050 SOUND 16,1
7060 SOUND 13,1
7070 NEXT
7080 G = G + 1
7090 IF G > 6 THEN 9000
7100 CP = CP + 64
7110 RETURN
8000 REM WRONG ANSWER PENALTY
8010 CD = CD - 100
8020 IF CD < 0 THEN CD = 0
8030 GOSUB 6000
8040 SOUND 2,5
8050 RETURN
9000 REM END ROUTINE
9010 PRINT@PO," ";
9020 IF CD => TS THEN GOSUB 10000
9030 PRINT@450,"PRESS <<RETURN>> FOR NEW GAME";
9035 K$ = INKEY$
9040 I$=INKEY$:IFI$=""THEN 9040
9050 IF ASC(I$) <> 13 THEN 9040
9060 GOSUB 35000
9070 GOSUB 31030
9080 CD = 6000 '(RESET COUNTDOWN)
9090 CP = 80 '(SET CURSOR POSITION)
9100 GOTO 310

```

```

10000 REM END ROUTINE
10010 IF TS = 0 THEN 10110
10020 PRINT@162, " ";
10030 SOUND 16,3
10040 SOUND 11,2
10050 SOUND 11,1
10060 SOUND 13,3
10070 SOUND 11,3
10080 SOUND 0,2
10090 SOUND 15,4
10100 SOUND 16,4
10110 TS = CD:TSS = N$:DN = 300
10120 RETURN
20000 REM SORT RANDOM SEQUENCE
20010 FOR N = 1 TO 6
20020 R=RND(NW)
20030 FOR NN = 1 TO N
20040 IF R=R(NN) THEN 20020
20050 NEXT NN
20060 R(N)=R
20070 R$(N)=W$(R(N))
20080 NEXT N
21000 REM SORT INTO A.O.
21010 FOR X = 1 TO 6
21020 MX=X
21030 FOR N = X+1 TO 6
21040 IF R$(N) < R$(MX) THEN MX = N
21050 NEXT N
21060 T$=R$(X)
21070 A$(X)=R$(MX)
21080 R$(MX)=T$
21090 NEXT X
21100 A$(6) = R$(MX)
21110 RETURN
30000 REM DRAW FRAME
30010 PRINT@0, " ";
30020 PRINT@430, " ";
30030 FOR N = 32 TO 448 STEP 32
30040 PRINT@N, " ";
30050 PRINT@N+31, " ";
30060 NEXT N
30070 PRINT@31, " ";
30080 POKE 29183,184
30090 RETURN
31000 REM INSTRUCTIONS
31010 CLS
31020 GOSUB 30000 '(FRAME)
31025 COLOR 3
31030 PRINT@70, " ";
31040 T$="(BY WAYNE G. RICHMOND)"
31050 FORN=1TOLEN(T$)
31060 PRINT@133,RIGHT$(T$,N);
31080 NEXT

```



```

31090 PRINT@194,"HOW QUICKLY CAN YOU SORT";
31100 PRINT@226,"A LIST OF WORDS INTO";
31110 PRINT@258,"ALPHABETICAL ORDER?";
31120 PRINT@322,"ARROW KEYS POSITION CURSOR";
31130 PRINT@354,"<<CTRL>> KEY SELECTS WORD";
31140 T$="MISTAKES COST 100 POINTS!"
31150 FOR N = 1 TO LEN(T$)
31160 PRINT@418,RIGHT$(T$,N);
31170 NEXT
31180 FOR N = 1 TO DN:NEXT
31195 K$=INKEY$
31200 PRINT@131," <<HIT ANY KEY TO BEGIN>>";
31205 SOUND 30,1
31210 FOR N = 1 TO 80
31220 I$=INKEY$:IFI$<>" THENRETURN
31230 NEXT
31240 PRINT@132," ";
31250 FOR N = 1 TO 20
31260 I$=INKEY$:IFI$<>" THENRETURN
31270 NEXT
31280 GOTO 31200
32000 REM INPUT NAME ROUTINE
32010 GOSUB 35000 '(CLEAR SCREEN)
32035 IF TS = 0 THEN 32060
32040 PRINT@130,;
32050 PRINTUSING"TOP SCORE: #,### % ";TS;TS$
32060 PRINT@226,"PLEASE TYPE YOUR FIRST NAME";
32070 PRINT@258,"AND PRESS <<RETURN>>";
32080 PRINT@354,;
32090 N$=""
32100 INPUT N$
32110 IF N$ = "" THEN 32000
32300 RETURN
35000 REM CLEAR CENTRE OF FRAME
35010 FOR N = 34 TO 450 STEP 32
35020 PRINT@N," ";
35030 NEXT
35040 RETURN
40000 REM WORD COLLECTION
40010 DATA ANT,DOG,CAT,HOUSE,MAN,BOAT,CAR,HEATER,CARPET,WATER
40020 DATA HAT,MOUSE,VAN,WIND,GREEN,WIDE,YELLOW,RED,BLACK,TOP
40030 DATA HAIR,HALF,HAIL,TICKET,THING,FOOT,FOLD,FOG,FOOD,FIT
40040 DATA PRINT,PRINCE,PRIDE,MUSIC,MUSTARD,SPACE,SPADE,FLOOR
40050 DATA SNUG,SOUND,SOUL,STAR,FLICK,STAGE,STABLE,CLEAN,FAT
40060 DATA WINDOW,WHEEL,WONDER,YOUNG,YOUR,YOUTH,ZOO,CAN,YELLOW


```

For explanation of graphics codes used, see next page.

EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

Line 1020 SHIFT + J
Line 10020 CTRL + INV + TEXT: NEW TOP SCORE
Line 30010 SHIFT + A, then SHIFT + Y's, then SHIFT + S
Line 30020 SHIFT + D, then SHIFT + T's, then SHIFT + F
Line 30040 SHIFT + I
Line 30050 SHIFT + U
Line 30070 SHIFT + S
Line 31030 CTRL + INV + TEXT: WORD RACE
Line 32110 SHIFT + U

What's My Number



Using this program, the VZ-200 will think of a number and your job is to try and guess what the number is. The computer will then inform as to whether your guess is "TOO HIGH" or "TOO LOW". You will also be given the opportunity set the upper limit for the number to be guessed. How many tries will YOU take to get it right?

FOR THE PROGRAMMER

This type of program is an excellent 'learning' program for the beginner and is the type of program that you can write for yourself. The program has deliberately been kept simple and we would encourage you to improve the program.

Note how line 3065 checks to see if the computer should print "GUESS" or "GUESSES".



```
10 REM * WHAT'S MY NUMBER? *
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (MAY 1983)
40 QS = CHR$(34)
50 COLOR 8,0
1000 REM INSTRUCTIONS
1010 CLS
1020 PRINT@39," "
1030 PRINT@71," "
1040 PRINT@103," "
1050 PRINT@161,"I WILL THINK OF A NUMBER AND"
1060 PRINT" YOU HAVE TO TRY TO GUESS IT."
1070 PRINT
1080 PRINT" I WILL SAY IF YOUR GUESS IS"
1090 PRINT
1100 PRINT" ";QS;"TOO LOW";QS;" OR ";QS;"TOO HIGH";QS;"."
1110 PRINT
1120 PRINT" WHAT IS MY UPPER LIMIT";
1130 INPUT R
1140 IF R = 0 THEN R = 100
1150 X = RND(R) '(SELECT NUMBER)
2000 REM INPUT GUESSES
2010 CLS
2020 PRINT
2030 PRINT" I AM THINKING OF A NUMBER"
2040 PRINT" BETWEEN 1 AND";R
2050 GN = GN + 1
2060 PRINT
2070 PRINTTAB(1)"GUESS #";GN;
2080 INPUT G
```

```

2100 IF G = X THEN 3000
2110 IF C < X THEN 4000
2120 IF G > X THEN 5000
3000 REM GUESS CORRECT
3010 CLS
3020 PRINT@39," "
3030 PRINT@71," "
3040 PRINT@103," "
3050 PRINT@160,G;"IS MY NUMBER AND IT ONLY"
3060 PRINT@193,"TOOK YOU";GN;"GUESS";
3065 IF GN = 1 THEN PRINT"!!" ELSE PRINT"ES!"
3070 PRINT:PRINT
3080 PRINT" RANGE: 1 -";R
3100 FOR N = 1 TO 10
3110 SOUND RND(30),1
3120 NEXT
3130 PRINT@385," "
3140 PRINT@417," "
3150 PRINT@449," "
3160 IS = INKEY$:IF IS <> " " THEN 3140
3170 RUN
4000 REM GUESS TOO LOW
4010 PRINT" "
4020 SOUND 1,3
4030 GOTO 2050
5000 REM GUESS TOO HIGH
5010 PRINT" "
5020 SOUND 30,3
5030 GOTO 2050

```



EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

- Line 1020 SHIFT + A, then SHIFT + Y's, then SHIFT + S
- Line 1030 CTRL + INV + TEXT: WHAT'S MY NUMBER ?
- Line 1040 SHIFT + D, then SHIFT + T's, then SHIFT + F
- Line 3020 SHIFT + A, then SHIFT + Y's, then SHIFT + S
- Line 3030 CTRL + INV + TEXT: CONGRATULATIONS!
- Line 3040 SHIFT + D, THEN SHIFT + T'S, THEN SHIFT + F
- Line 3130 SHIFT + A, then SHIFT + Y's, then SHIFT + S
- Line 3140 CTRL + INV + TEXT: PRESS (SPACE) FOR NEW GAME
- Line 3150 SHIFT + D, then SHIFT + T's, then SHIFT + F
- Line 4010 CTRL + INV + TEXT: TOO LOW!
- Line 5010 CTRL + INV + TEXT: TOO HIGH!

What's My Letter *

If you think that this program is very similar to the preceding one, you're quite correct. The difference here is that this time you have to guess a letter from the alphabet rather than a letter.

FOR THE PROGRAMMER

This is a good example of how to 'adapt' an existing program rather than starting completely from scratch to produce a 'new' version of the program. We suggest that you compare the previous program in order to determine where the changes have been made and see how easily this program has been adapted.

```
10 REM * WHAT'S MY LETTER? *
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (MAY 1983)
40 Q$ = CHR$(34)
50 COLOR 8,0
1000 REM INSTRUCTIONS
1010 CLS
1020 PRINT@39," "
1030 PRINT@71," "
1040 PRINT@103," "
1050 PRINT@162,"I WILL THINK OF A LETTER AND"
1060 PRINT@194,"YOU HAVE TO TRY TO GUESS IT."
1070 PRINT
1080 PRINT" I WILL SAY IF YOUR GUESS IS"
1090 PRINT
1100 PRINT" ";Q$;"TOO LOW ";Q$;" OR ";Q$;"TOO HIGH";Q$;". "
1110 PRINT
1120 PRINT" ABCDEFGHIJKLMNOPQRSTUVWXYZ"
1130 PRINT@451," "
1135 K$ = INKEY$
1140 I$ = INKEY$:IF I$ <> " " THEN 1140
1150 X = RND(26) + 64 '(SELECT LETTER)
2000 REM INPUT GUESSES
2010 CLS
2020 PRINT
2030 PRINT" I AM THINKING OF A LETTER"
2040 PRINT" BETWEEN A AND Z"
2050 GN = GN + 1
2060 PRINT
2070 PRINTTAB(1)"GUESS #";GN;
2080 INPUT G$
2085 IF G$ < "A" OR G$ > "Z" THEN 2080
2090 G = ASC(G$)
```

```

2100 IF G = X THEN 3000
2110 IF G < X THEN 4000
2120 IF G > X THEN 5000
3000 REM GUESS CORRECT
3010 CLS
3020 PRINT@39, " "
3030 PRINT@71, " "
3040 PRINT@103, " "
3050 PRINT@161,Q$;CHR$(G);Q$;" IS MY LETTER AND IT ONLY"
3060 PRINT@193,"TOOK YOU";GN;"GUESS";
3065 IF GN = 1 THEN PRINT"!!" ELSE PRINT"ES!"
3070 PRINT:PRINT
3100 FOR N = 1 TO 10
3110 SOUND RND(30),1
3120 NEXT
3130 PRINT@385, " "
3140 PRINT@417, " "
3150 PRINT@449, " "
3160 IS$ = INKEY$:IF IS$ <> " " THEN 3140
3170 RUN
4000 REM GUESS TOO LOW
4010 PRINT " "
4020 SOUND 1,3
4030 GOTO 2050
5000 REM GUESS TOO HIGH
5010 PRINT " "
5020 SOUND 30,3
5030 GOTO 2050

```



EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

Line 1020	SHIFT + A, then SHIFT + Y's, then SHIFT + S
Line 1030	CTRL + INV + TEXT: WHAT'S MY LETTER ?
Line 1040	SHIFT + D, then SHIFT + T's, then SHIFT + F
Line 1130	CTRL + INV + TEXT: PRESS (SPACE) TO BEGIN
Line 3020	SHIFT + A, then SHIFT + Y's, then SHIFT + S
Line 3030	CTRL + INV + TEXT: CONGRATULATIONS !
Line 3040	SHIFT + D, then SHIFT + T's, then SHIFT + F
Line 3130	SHIFT + A, then SHIFT + Y's, then SHIFT + S
Line 3140	CTRL + INV + TEXT: PRESS (SPACE) FOR NEW GAME
Line 3150	SHIFT + D, then SHIFT + T's, then SHIFT + F
Line 4010	CTRL + INV + TEXT: TOO LOW !
Line 5010	CTRL + INV + TEXT: TOO HIGH !

OPQRS
NOPQRS

VZ-200 Maths Quiz

This program lets you practice your maths skills. You are given a choice of addition, subtraction, multiplication, division or a mixture of these at any of three levels of difficulty. When you decide to quit, you will be presented with a complete report. By the way, your VZ-200 makes random comments after each effort. It can in fact be quite rude!

FOR THE PROGRAMMER

Note the interesting method used in line 160 to select the required choice from the menu. Note also how each section of the program is separated by REM statements from line 1000 to allow you easy access to follow the logic.

```
10 REM * VZ-200 MATHS QUIZ *
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (MAY 1983)
40 CLS
50 CLEAR 200
60 FOR N = 1 TO 10:READ C$(N):NEXT '(READ CORRECT WORDS)
70 FOR N = 1 TO 10:READ W$(N):NEXT '(READ WRONG WORDS)
100 GOSUB 20000 '(DRAW FRAME)
110 PRINT@194,"1.  ADDITION";
120 PRINT@226,"2.  SUBTRACTION";
130 PRINT@258,"3.  MULTIPLICATION";
140 PRINT@290,"4.  DIVISION";
145 PRINT@322,"5.  MIXED";
150 PRINT@386,"WHICH TYPE OF MATHS (1-5)";
155 SOUND 30,1
160 INPUT T$:T$ = LEFT$(T$,1):IF T$ = "" THEN 150
170 IF T$ < "1" OR T$ > "5" THEN PRINT@413,"  "":GOTO 150
175 SOUND 30,1:GOSUB 500
180 PRINT@418,"WHICH LEVEL (1-3)";
190 INPUT L$:L$ = LEFT$(L$,1):IF L$ = "" THEN 180
200 IF L$ < "1" OR L$ > "3" THEN PRINT@437,"  "":GOTO 180
205 SOUND 30,1
210 L = VAL(L$)
220 GOSUB 21000 '(CLEAR SCREEN)
250 PRINT@66,"          ";L$;" (";T$;")          ";
260 COLOR 7
270 PRINT@421,"          ";
280 GOTO 1000
500 IF T$ = "1" THEN T$$ = "ADD.":TL$ = "ADDITION"
510 IF T$ = "2" THEN T$$ = "SUB.":TL$ = "SUBTRACTION"
520 IF T$ = "3" THEN T$$ = "MULT.":TL$ = "MULTIPLICATION"
530 IF T$ = "4" THEN T$$ = "DIV.":TL$ = "DIVISION"
540 IF T$ = "5" THEN T$$ = "MIXED":TL$ = "MIXED"
550 RETURN
```


```

1000 REM PRESENT QUESTION
1010 PRINT@257,"";
1020 K = K + 1 '(QUESTION COUNTER)
1025 PRINT@91,,:PRINTUSING"###";K
1030 IF T$ = "5" THEN T = RND(4) ELSE T = VAL(T$)
1040 IF T = 1 THEN GOSUB 2000
1050 IF T = 2 THEN GOSUB 3000
1060 IF T = 3 THEN GOSUB 4000
1070 IF T = 4 THEN GOSUB 5000
1100 PRINT@257,A;S$;B;"=";
1110 INPUT X$:IF LEFT$(X$,1) = "Q" THEN 6000' (CODA)
1120 X = VAL(X$)
1130 IF X = C THEN GOSUB 1200 ELSE GOSUB 1300
1140 GOTO 1010
1200 REM ANSWER CORRECT
1210 CT = CT + 1 '(CORRECT ANSWER COUNTER)
1220 PRINT@275,C#(RND(10));"!!!";
1230 GOSUB 1500
1240 FOR N = 1 TO 4
1250 SOUND RND(10) + 20,1
1260 NEXT
1270 RETURN
1300 REM ANSWER WRONG
1310 PRINT@275,W#(RND(10));"!!!";
1320 GOSUB 1500
1330 SOUND 1,7
1340 RETURN
1500 REM PRINT STATISTICS
1510 PRINT@66,"###";L$;" (";T$;")";
1540 PRINT@130,"###";
1550 PRINTUSING"###%";CT/K*100
1560 RETURN
2000 REM ADDITION
2010 S$ = "+"
2020 IF L = 1 THEN A = RND(10):B = RND(10)
2030 IF L = 2 THEN A = RND(10) + 40:B = RND(20)
2040 IF L = 3 THEN A = RND(60) + 60:B = RND(60) + 60
2050 C = A + B:RETURN
3000 REM SUBTRACTION
3010 S$ = "-"
3020 IF L = 1 THEN C = RND(10):B = RND(10)
3030 IF L = 2 THEN C = RND(10) + 40:B = RND(20)
3040 IF L = 3 THEN C = RND(60) + 60:B = RND(60) + 60
3050 A = C + B:RETURN
4000 REM MULTIPLICATION
4010 S$ = "X"
4020 IF L = 1 THEN A = RND(10):B = RND(10)
4030 IF L = 2 THEN A = RND(10):B = RND(30) + 10
4040 IF L = 3 THEN A = RND(30) + 20:B = RND(50) + 50
4050 C = A * B: RETURN
5000 REM DIVISION
5010 S$ = "/"

```


EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

Line 160 SHIFT + U
 Line 190 SHIFT + U
 Line 250 CTRL + INV + TEXT: LEVEL: QUEST:
 Line 270 CTRL + INV + TEXT: (ANSWER "Q" TO QUIT)
 Line 1220 CTRL + INV + TEXT: !
 Line 1310 CTRL + INV + TEXT: !
 Line 1510 CTRL + INV + TEXT: LEVEL: QUEST:
 Line 1540 CTRL + INV + TEXT: CORRECT:
 Line 6020 CTRL + INV + TEXT: TYPE:
 Line 6030 CTRL + INV + TEXT: # ATTEMPTS:
 Line 6040 CTRL + INV + TEXT: # CORRECT:
 Line 6050 CTRL + INV + TEXT: # WRONG:
 Line 6060 CTRL + INV + TEXT: PERCENTAGE:
 Line 6070 CTRL + INV + TEXT: PRESS (SPACE) FOR NEW QUIZ
 Line 20020 SHIFT + A, then SHIFT + Y's, then SHIFT + S
 Line 20030 SHIFT + D, then SHIFT + T's, then SHIFT + F
 Line 20050 SHIFT + I
 Line 20060 SHIFT + U
 Line 20080 SHIFT + S
 Line 20120 CTRL + INV + TEXT: VZ-200 MATHS QUIZ
 Line 30000 CTRL + INV + TEXT: SPOT ON, TERRIFIC, BEAUTY,
 FANTASTIC, BINGO
 Line 30010 CTRL + INV + TEXT: MARVELLOUS, BULLS EYE, GREAT,
 WONDERFUL, WOW
 Line 30020 CTRL + INV + TEXT: TOO BAD, GOOD TRY, NO WAY,
 WRONG, NOT RIGHT
 Line 30030 CTRL + INV + TEXT: BOMBED, BOMB OUT, OFF OFF,
 FIZZER, NEVER MIND



A B C D E F G H I J K L M N O P Q R S T U V W X Y Z
- - - - -

VZ-200 Music Box

In this program, your VZ-200 indulges in a little nationalistic nostalgia. It will with great dexterity and skill, play 4 of Australia's best known tunes:

The Pub With No Beer

Advance Australia Fair

Botany Bay

Tie Me Kangaroo Down.

FOR THE PROGRAMMER

Because this program 'plays' four different tunes, each of the required sets of DATA statements has been separated and is treated individually once selected from the menu. This has been done to allow you to see how each tune has been derived. Note the DIM statements in lines 100-130 which sets the size (number) of DATA statements used for each tune. Note also the simple way in which each tune is played by using a FOR...NEXT loop such as in lines 1010-1070.

```
10 REM * VZ-200 MUSIC BOX *
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (MAY 1983)
100 DIM AP(47),AD(47)
110 DIM BP(70),BD(70)
120 DIM CP(41),CD(41)
130 DIM DP(33),DD(33)
140 GOTO 600
200 REM READ DATA
205 PRINT@418,"PLEASE WAIT . . . .";
210 FOR N = 1 TO 47
220   READ AP(N):READ AD(N)
230 NEXT
240 FOR N = 1 TO 70
250   READ BP(N):READ BD(N)
260 NEXT
270 FOR N = 1 TO 41
280   READ CP(N):READ CD(N)
290 NEXT
300 FOR N = 1 TO 33
310   READ DP(N):READ DD(N)
320 NEXT
330 SOUND 30,1
340 RETURN
```




Note

```

600 REM MENU
610 CLS
620 COLOR 8,0
630 PRINT@0," ";
640 PRINT@480," ";
650 FOR N = 32 TO 448 STEP 32
660 PRINT@N," ";
670 PRINT@N+31," ";
680 NEXT
690 PRINT@31," ";
700 POKE 29183,248
710 COLOR 7
720 PRINT@71," ";
730 T$ = "(BY WAYNE G. RICHMOND)"
740 FOR N = 1 TO LEN(T$)
750 PRINT@133,RIGHT$(T$,N);
755 FORX=1TO4:NEXT
760 NEXT
770 PRINT@226,"1. PUB WITH NO BEER";
780 PRINT@258,"2. ADVANCE AUSTRALIA FAIR";
790 PRINT@290,"3. BOTANY BAY";
800 PRINT@322,"4. TIE ME KANGAROO DOWN";
805 GOSUB 200
810 PRINT@418,"TYPE 1, 2, 3 OR 4 ";
820 K$ = INKEY$
830 I$ = INKEY$:IF I$ = "" THEN 830
840 IF I$ = "1" THEN 1000
850 IF I$ = "2" THEN 2000
860 IF I$ = "3" THEN 3000
870 IF I$ = "4" THEN 4000
880 GOTO 830
1000 REM PUB WITH NO BEER
1010 PRINT@418," ";
1020 FOR R = 1 TO 2
1030 FOR N = 1 TO 47
1040 SOUND AP(N),AD(N)
1050 NEXT N
1060 SOUND 0,1
1070 NEXT R
1080 GOTO 810
2000 REM ADVANCE AUSTRALIA FAIR
2010 PRINT@418," ";
2020 FOR N = 1 TO 70
2030 SOUND BP(N),BD(N)
2040 NEXT N
2050 GOTO 810
3000 REM BOTANY BAY
3010 PRINT@418," ";
3020 FOR R = 1 TO 2
3030 FOR N = 1 TO 41
3040 SOUND CP(N),CD(N)
3050 NEXT N
3060 SOUND 0,1
3070 NEXT R
3080 GOTO 810

```



```

4000 REM TIE ME KANGAROO DOWN
4010 PRINT@418, "
4020 FOR N = 1 TO 32
4030 SOUND DP(N),DD(N)
4040 NEXT N
4050 FOR N = 1 TO 26
4060 SOUND DP(N),DD(N)
4070 NEXT N
4080 SOUND DP(33),DD(33)
4090 GOTO 810
10000 REM DATA FOR "PUB WITH NO BEER"
10010 DATA 18,2,18,4,17,4,20,4,18,4,13,5,10,2,8,4,7,4,8,4
10020 DATA 15,6,15,2,15,2,13,4,17,4,15,4,13,6
10030 DATA 8,2,8,2,8,4,10,4,11,4,10,6
10040 DATA 18,2,18,2,18,4,17,4,20,4,18,4,13,5,10,2,8,4,7,4,8,4
10050 DATA 15,6,15,2,15,2,13,4,17,4,15,4,13,6
10060 DATA 8,2,10,2,11,4,10,4,8,4,6,6
20000 REM DATA FOR "ADVANCE AUSTRALIA FAIR"
20010 DATA 11,4,16,4,11,4,8,4,11,4,16,5,16,2,16,4
20020 DATA 20,4,18,4,16,4,15,4,16,4,18,8
20030 DATA 11,4,16,4,11,4,8,4,4,4,11,5,11,2,11,4
20040 DATA 20,4,18,4,16,4,15,4,13,4,11,8
20050 DATA 11,4,13,5,15,2,16,4,13,4,11,5,8,2,8,4
20060 DATA 11,4,13,4,16,4,21,4,20,4,18,8
20070 DATA 11,4,13,5,15,2,16,4,13,4,11,5,16,2,16,4,18,4
20080 DATA 20,5,16,2,18,5,15,2,16,8
20090 DATA 20,4,21,4,20,4,18,4,16,4,15,4,13,4,11,4
20100 DATA 16,4,20,5,16,2,18,5,15,2,16,8
30000 REM DATA FOR "BOTANY BAY"
30010 DATA 7,1,9,1,11,2,14,2,14,2,9,2,12,2,12,2,11,1,9,1,7,5,0,1
30020 DATA 14,2,11,2,14,2,19,2,12,2,16,2,19,2,14,6,0,1
30030 DATA 14,2,19,2,18,2,19,2,21,2,19,2,16,2,14,1,11,1,7,5,0,1
30040 DATA 7,1,9,1,11,2,14,2,14,2,9,2,12,2,11,2,7,6
40000 REM DATA FOR "TIE ME KANGAROO DOWN"
40010 DATA 4,4,9,2,13,4,11,2,9,4,6,6,11,6
40020 DATA 4,4,8,2,11,4,14,2,18,4,16,8,0,1
40030 DATA 13,4,16,2,13,4,11,2,9,4,6,6,11,6
40040 DATA 4,4,8,2,11,4,9,2,8,4,9,4
40050 DATA 16,2,16,1,16,2,16,1,16,4
40060 DATA 9,7

```

EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

- Line 630 SHIFT + A, then SHIFT + Y's, then SHIFT + S
- Line 640 SHIFT + D, then SHIFT + T's, then SHIFT + F
- Line 660 SHIFT + I
- Line 670 SHIFT + U
- Line 690 SHIFT + S
- Line 720 CTRL + INV + TEXT: VZ-200 MUSIC BOX
- Line 1010 CTRL + INV + TEXT: THE PUB WITH NO BEER
- Line 2010 CTRL + INV + TEXT: ADVANCE AUSTRALIA FAIR
- Line 3010 CTRL + INV + TEXT: BOTANY BAY
- Line 4010 CTRL + INV + TEXT: TIE ME KANGAROO DOWN

VZ-200 Clock *

This program turns your VZ-200 into a digital timepiece. In order to maintain the accuracy of the program, we suggest that you key in the program EXACTLY as it is - spaces, REM's and all!

FOR THE PROGRAMMER

The loop in line 3030 is used simply to act as a delay to adjust the clock's timekeeping accuracy. Adjustment of the period of delay can be made by altering the value (by either increasing or decreasing the value) of TW in line 80. Another method used to 'slow things' down would be to add REM statements between lines 3010 and 3060. You'd be surprised to learn just how much time they can take up in a program.

Note also the testing done from line 7000 to ensure that the time figures that have been entered, are valid.

He went on
a diet

```
10 REM * VZ-200 CLOCK *
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (MAY 1983)
50 CLS
60 GOSUB 6000 '(DRAW SCREEN FRAME)
70 COLOR 1,1
80 TW = 742 '(SET TIME WASTER)
1000 REM DRAW CLOCK FRAME
1010 PRINT@168,".....";
1020 FOR N = 200 TO 264 STEP 32
1030 PRINT@N,"   ";
1040 NEXT
1050 PRINT@296,".....";
2000 REM INITIALISE TIME
2010 PRINT@108,".....";
2020 PRINT@354,".....";
2030 PRINT@366,".....";
2040 INPUT H$
2050 X$ = H$:GOSUB 7000 '(CHECK VALIDITY)
2060 IF X$ = "" THEN 2030
2070 H = VAL(H$)
2080 PRINT@234,,:PRINTUSING"##";H
2090 PRINT@366,".....";
2100 INPUT M$
2110 X$ = M$:GOSUB 8000 '(CHECK VALIDITY)
2120 IF X$ = "" THEN 2090
2130 PRINT@375," ";
2140 M = VAL(M$)
```



```

2150 IF M < 10 THEN F$ = "0#" ELSE F$ = "##"
2160 PRINT@239,,:PRINTUSING F$;M
2170 PRINT@366,"#=#";
2180 INPUT S$
2190 X$ = S$:GOSUB 8000' (CHECK VALIDITY)
2200 IF X$ = "" THEN 2170
2210 S = VAL(S$)
2220 IF S < 10 THEN F$ = "0#" ELSE F$ = "##"
2230 PRINT@244,,:PRINTUSING F$;S
2240 PRINT@354,"";
2250 PRINT@419,"";
2260 K$ = INKEY$
2270 I$ = INKEY$:IF I$ <> " " THEN 2270
2280 PRINT@419,"";
2290 GOTO 3030
3000 REM TIMING ROUTINE
3010 GOSUB 4000' (PRINT TIME)
3030 FOR N = 1 TO TW:NEXT
3060 GOSUB 5000' (ADJUST TIME)
3070 GOTO 3010
4000 REM PRINT TIME
4010 PRINT@234,;
4020 IF M < 10 AND S < 10 THEN 4160
4030 IF M < 10 THEN 4140
4040 IF S < 10 THEN 4120
4100 PRINTUSING"## ## ##";H;M;S
4110 RETURN
4120 PRINTUSING"## ## 0#";H;M;S
4130 RETURN
4140 PRINTUSING"## 0# ##";H;M;S
4150 RETURN
4160 PRINTUSING"## 0# 0#";H;M;S
4170 RETURN
5000 REM UPDATE TIME
5010 S = S + 1:IF S < 60 THEN RETURN
5015 S = 0
5020 M = M + 1:IF M < 60 THEN RETURN
5025 M = 0
5030 H = H + 1:IF H < 13 THEN RETURN
5040 H = 1:RETURN
6000 REM DRAW SCREEN FRAME
6010 COLOR 5,1
6020 PRINT@0,"";
6030 PRINT@480,"";
6040 FOR N = 32 TO 448 STEP 32
6050 PRINT@N,"";
6060 PRINT@N+31,"";
6070 NEXT
6080 PRINT@31,"";
6090 POKE 29183,200
6100 RETURN

```

```

7000 REM CHECK VALIDITY OF INPUT (HOUR)
7010 X = VAL(X$)
7020 IF X < 1 OR X > 12 THEN X$ = ""
7030 IF X$ = "" THEN X$ = ""
7040 IF X$ = "" THEN PRINT@372," ";
7050 RETURN
8000 REM CHECK VALIDITY OF INPUT (MINUTES & SECONDS)
8010 X = VAL(X$)
8020 IF X < 0 OR X > 59 THEN X$ = ""
8030 IF X$ = "" THEN X$ = ""
8040 IF X$ = "" THEN PRINT@374," ";
8050 RETURN

```



EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

```

Line 1010 SHIFT + Y's
Line 1030 SHIFT + J's
Line 1050 SHIFT + T's
Line 2010 CTRL + INV + TEXT: VZ-200
Line 2020 CTRL + INV + TEXT: SET CLOCK:
Line 2030 CTRL + INV + TEXT: HOUR:
Line 2090 CTRL + INV + TEXT: MINUTE:
Line 2170 CTRL + INV + TEXT: SECOND:
Line 2250 CTRL + INV + TEXT: PRESS (SPACE) TO START
Line 4100 SHIFT + J's
Line 4120 SHIFT + J's
Line 4140 SHIFT + J's
Line 4160 SHIFT + J's
Line 6020 SHIFT + Y's
Line 6030 SHIFT + T's
Line 6050 SHIFT + I
Line 6060 SHIFT + U
Line 6080 SHIFT + S
Line 7030 SHIFT + U
Line 8030 SHIFT + U

```



What Day Of The Week Is It

This neat little program will calculate which day of the week it is (or will be) for virtually any date within this century. What day of the week were you born on?

It is very important to ensure that the date is entered correctly, i.e. DD/MM/YY. (e.g. 12/08/47 for 12th August, 1947). To reiterate, two digits are required for each section of the date.

FOR THE PROGRAMMER

Note the manipulation of string data (the date in this case) in lines 1160-1260 used to extract the day, month and year from the INPUTed string DD/MM/YY using the string functions LEFT\$, RIGHT\$, MID\$ and VAL.



```
10 REM * WHAT DAY OF THE WEEK IS IT?*
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (MAY 1983)
40 COLOR ,0
1000 REM INPUT DATE
1010 CLS
1020 GOSUB 10000 '(DRAW FRAME)
1030 COLOR 7
1040 PRINT@69,"          ";
1050 T$="(BY WAYNE G. RICHMOND)"
1060 FOR N = 1 TO LEN(T$)
1070   PRINT@133,RIGHT$(T$,N);
1080 NEXT
1090 PRINT@194,"THIS PROGRAM CAN CALCULATE";
1100 PRINT@226,"WHAT DAY OF THE WEEK ANY";
1110 PRINT@258,"PAST OR FUTURE DATE FALLS ON!";
1120 PRINT@322,"TYPE IN THE DATE (DD/MM/YY)";
1130 PRINT@354,"E.G. 07/05/83 (7TH MAY 1983)";
1140 PRINT@418,;
1150 INPUT T$
1155 IF ASC(T$) = 138 THEN 1000
1160 T$ = LEFT$(T$,8)
1200 Y = VAL(RIGHT$(T$,2))
1210 M = VAL(MID$(T$,4,2))
1220 IF M > 12 THEN 1000
1230 D = VAL(LEFT$(T$,2))
1240 IF D > 31 THEN 1000
1250 IF MID$(T$,3,1) > "/" THEN 1000
1260 IF MID$(T$,6,1) > "/" THEN 1000
1550 IF INT(Y/4) = Y/4 THEN 3000 '(CHECK FOR LEAP YEAR)
```




```
10040 FOR N = 32 TO 448 STEP 32
10050 PRINTEN,"#";
10060 PRINTEN+31,"#";
10070 NEXT
10080 PRINTEN31,"#";
10090 POKE 29183,248
10100 RETURN
10250 IF INT(Y/4) = Y/4 THEN 12000 '(CHECK FOR LEAP YEAR)
```

EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

- Line 1040 CTRL + INV + TEXT: WHAT DAY OF THE WEEK ?
- Line 10020 SHIFT + A, then SHIFT + Y's
- Line 10030 SHIFT + D, then SHIFT + T's
- Line 10050 SHIFT + I
- Line 10060 SHIFT + U
- Line 10080 SHIFT + S

Crystal Spectacular *


Here, we have allowed the VZ-200 to show its artistic side in this stunningly beautiful program. Random rectangular patterns are drawn all over the screen producing a brilliant colour display. At any time, you can freeze the display by simply holding the SPACE key until you hear a low buzz. You may continue the program simply by hitting the RETURN key. With this program running, you are guaranteed to impress your wife, kids and friends and (at last) justify your purchase of the VZ-200.

FOR THE PROGRAMMER

In this program, High-resolution graphics is used by switching display mode at line 60. Line 300 randomly chooses the new locations (X and Y) and size (LT) of the next 'crystal'.

Line 310 checks to see if the SPACE key is being held down. If it is, the program moves to the routine beginning at line 400 where it is held at the INKEY\$ in line 410 until it detects a RETURN key (ASCII 13).

```
10 REM * CRYSTAL SPECTACULAR *
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (MAY 1983)
40 GOSUB 1000
50 CLS
60 MODE (1)
70 COLOR 8,1
100 GOSUB 300
110 FOR X = X TO X + I
120   SET (X,Y)
130 NEXT X
140 FOR Y = Y TO Y + I
150   SET (X,Y)
160 NEXT Y
170 I = I + 1:IF I > LT THEN 100
180 FOR X = X TO X - I STEP -1
190   SET (X,Y)
200 NEXT X
210 FOR Y = Y TO Y - I STEP -1
220   SET (X,Y)
230 NEXT Y
240 I = I + 1:IF I > LT THEN 100
250 COLOR RND(4)+4
260 GOTO 110
300 X=RND(67)+27:Y=26+RND(6):LT=RND(48)+5:I=0
310 K$ = INKEY$:I$ = INKEY$:IF I$ = " " THEN 400 ELSE RETURN
400 SOUND 10,3
410 I$ = INKEY$:IF I$ = "" THEN 410
420 IF ASC(I$) = 13 THEN SOUND 30,1:RETURN ELSE 410
```



```

1000 REM INTRODUCTION
1010 CLS:MODE (0)
1020 COLOR 8,0
1030 PRINT@0," ";
1040 PRINT@480," ";
1050 FOR N = 32 TO 448 STEP 32
1060 PRINT@N," ";
1070 PRINT@N+31," ";
1080 NEXT
1090 PRINT@31," ";
1100 POKE 29183,248
1105 COLOR 7
1110 PRINT@59," ";
1120 T$ = "(BY WAYNE G. RICHMOND)"
1130 FOR N = 1 TO LEN(T$)
1140 PRINT@133,RIGHT$(T$,N);
1150 NEXT
1160 PRINT@194,"THIS PROGRAM CREATES AMAZING";
1170 PRINT@226,"COLOUR CRYSTALS!";
1180 PRINT@290,"HOLD DOWN <<SPACE>> TO PAUSE";
1190 PRINT@322,"PRESS <<RETURN>> TO CONTINUE";
1200 PRINT@419," ";
1205 K$ = INKEY$
1210 I$ = INKEY$:IF I$ = "" THEN 1210
1220 IF ASC(I$) <> 13 THEN 1210 ELSE SOUND 30,1:RETURN

```

EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

Line 1030	SHIFT + A, then SHIFT + Y's
Line 1040	SHIFT + D, then SHIFT + T's
Line 1060	SHIFT + I
Line 1070	SHIFT + U
Line 1090	SHIFT + S
Line 1110	CTRL + INV + TEXT: CRYSTAL SPECTACULAR
Line 1200	CTRL + INV + TEXT: HIT (RETURN) TO START !

Write A Story

This program creates absolutely hilarious stories! There are two stories in memory with a number of words left out. The computer will ask you to put in the missing words and the computer will then print the story for you.

If you have a printer attached the program can produce a copy of the story for you on paper.

One of the stories predicts the future for you and your friends while the other extols the virtues of "Good Health"!

To help you in answering the questions, here is a quick cram in English grammar:-

An ADJECTIVE describes something or somebody, e.g. lumpy, ticklish, blonde, messy etc.

A NOUN is the name of a person, place or thing, e.g. woman, footpath, dog, book, umbrella etc.

A PLURAL NOUN is a noun naming more than one person, e.g. women, yards etc.

An EXCLAMATION is any sort of odd sound or outcry, e.g. gasp, ouch, wow, gadzooks etc.

N.B. Try to think of unusual and/or funny words to use. Don't make them too long or they won't fit the screen properly. If you want a real good laugh, this is it!

FOR THE PROGRAMMER


Rather than using DATA statements as in previous programs to reproduce information, it was decided to use string functions and show you how they can be manipulated. The manipulation of string data is one of the many useful advantages that BASIC has to offer. No other language allows the freedom (of BASIC) in this one respect.

As this program does a fair bit of 'jumping around' within itself, we suggest that you take a particular choice from the menu, and using a separate pad and pencil, write down the 'flow' of the program by following the logic as it goes both up and down within the program.

```

10 REM ** WRITE A STORY **
20 REM BY WAYNE G. RICHMOND
50 CLEAR200:Y$=CHR$(34):COLOR 8,0
70 L.L.#=".....":GOTO1000
400 CLS:PRINTLL$;
430 PRINT" PLEASE PROVIDE THE REQUIRED"
440 PRINT" WORD BY TYPING IT AND PRESSING"
450 PRINT" THE <<RETURN>> KEY."
460 PRINTLL$;:RETURN
500 PRINT@194,Q$
520 PRINT
530 INPUT Z$
540 IFZ$=""THEN500
550 PRINT@194,"
560 PRINT@256,"
"
":RETURN
1000 CLS:PRINT@40,"
1100 PRINT@162,"1. FUTURE PROFESSION
1110 PRINT@258,"2. GOOD HEALTH
1120 PRINT@418,"TYPE 1 OR 2 ";
1130 I$=INKEY$:IFI$=""THEN1130
1140 IFI$="1"THEN2000ELSEIFI$="2"THEN5000ELSE1130
2000 GOSUB 400
2010 Q$="ADJECTIVE":GOSUB500:A$=Z$
2015 Q$="NAME OF FEMALE":GOSUB500:B$=Z$
2020 Q$="VERB ENDING IN "+Y$+"ING"+Y$:GOSUB500:C$=Z$
2030 Q$="A PROFESSION OR JOB":GOSUB500:D$=Z$
2040 Q$="A FEMALE CELEBRITY":GOSUB500:E$=Z$
2050 Q$="NAME OF MALE":GOSUB500:F$=Z$
2060 Q$="A MALE CELEBRITY":GOSUB500:G$=Z$
2065 Q$="NAME OF PERSON":GOSUB500:H$=Z$
2070 Q$="VERB ENDING IN "+Y$+"ING"+Y$:GOSUB500:I$=Z$
2080 Q$="PLURAL NOUN":GOSUB500:J$=Z$
2090 Q$="NOUN":GOSUB500:K$=Z$
2100 Q$="NAME OF PERSON":GOSUB500:L$=Z$
3000 CLS:PRINT@33,"
3030 PRINT@64,
3040 PRINT"I HAVE NOW PROCESSED
3050 PRINT"RESULTS OF YOUR ";A$
3060 PRINT"APTITUDE TESTS. I OFFICIALLY"
3070 PRINT"PREDICT THAT ";B$
3080 PRINT"HAS A LOT OF TALENT FOR"
3090 PRINTC$;". SHE WILL"
3100 PRINT"PROBABLY BECOME A FAMOUS"
3110 PRINTD$;" LIKE ";E$;". "
3115 GOSUB 11000:PRINT
3120 PRINTF$;" WILL BECOME A"
3130 PRINT"SECOND ";G$;". ON"
3140 PRINT"THE OTHER HAND, ";H$
3150 PRINT"SHOULD TAKE UP ";I$
3160 PRINT"AND GET A JOB HANDLING"
3170 PRINTJ$;". "
3190 PRINT"WHEN I GROW UP I WANT TO BE"
3200 PRINT"EITHER A FIREMAN, A DOCTOR, OR"
3210 PRINT"A";:X$=K$:GOSUB8000:PRINTW$;

```



```

3215 PRINT " ;K$;". IN THE MEANTIME,"
3220 PRINT"I AM MODELLING MYSELF AFTER MY"
3230 PRINT"HERO, ";L$;". "
3240 PRINT
3250 PRINT"
3260 GOSUB12000
3270 IFPY$="R"THEN3000
4000 LPRINTTAB(9)"FUTURE PROFESSIONS"
4030 LPRINT:LPRINT
4040 LPRINT"I HAVE NOW PROCESSED ALL THE ";
4050 LPRINT"RESULTS OF YOUR ";A$
4060 LPRINT"APTITUDE TESTS. I OFFICIALLY ";
4070 LPRINT"PREDICT THAT ";B$
4080 LPRINT"HAS A LOT OF TALENT FOR ";
4090 LPRINTC$;". SHE WILL"
4100 LPRINT"PROBABLY BECOME A FAMOUS ";
4110 LPRINTD$;" LIKE ";E$;". "
4120 LPRINTF$;" WILL BECOME A SECOND ";G$;". ON"
4140 LPRINT"THE OTHER HAND, ";H$;
4150 LPRINT" SHOULD TAKE UP ";I$
4160 LPRINT"AND GET A JOB HANDLING ";
4170 LPRINTJ$;". "
4180 LPRINT
4190 LPRINT"WHEN I GROW UP I WANT TO BE ";
4200 LPRINT"EITHER A FIREMAN, A DOCTOR, OR"
4210 LPRINT"A";:X$=K$:GOSUB8000:LPRINTW$;" ";K$;". IN THE ";
4220 LPRINT"MEANTIME, I AM MODELLING MYSELF"
4230 LPRINT"AFTER MY HERO, ";L$;". "
4240 LPRINT
4250 LPRINT"*****";
4260 LPRINT"*****"
4270 LPRINT:LPRINT
4280 GOSUB 12000
4290 IFPY$="R"THEN3000
4300 GOTO4000
5000 GOSUB400
5010 Q$="PLURAL NOUN":GOSUB500:A$=Z$
5020 Q$="NOUN":GOSUB500:B$=Z$
5030 Q$="A FOREIGN WORD":GOSUB500:C$=Z$
5040 Q$="ANOTHER FOREIGN WORD":GOSUB500:D$=Z$
5050 Q$="NONSENSE WORD":GOSUB500:E$=Z$
5060 Q$="A SILLY WORD":GOSUB500:F$=Z$
5070 Q$="NOUN":GOSUB 500:G$=Z$
5075 Q$="ANOTHER NOUN":GOSUB500:H$=Z$
5080 Q$="ADJECTIVE":GOSUB500:I$=Z$
5090 Q$="ANOTHER ADJECTIVE":GOSUB500:J$=Z$
5100 Q$="NOUN":GOSUB500:K$=Z$
5110 Q$="AN EXCLAMATION":GOSUB500:L$=Z$
6000 CLS
6020 PRINT@37," ***** "
6030 PRINT@64,
6040 PRINT"GOOD HEALTH IS MORE IMPORTANT"
6050 PRINT"THAN HAVING A LOT OF"

```



```

6060 PRINTA$;" OR BEING A VERY"
6070 PRINT"IMPORTANT ";B$;". "
6080 PRINT"THE OLD GREEKS USED TO"
6090 PRINT"SAY: ";Y$;C$;" ";D$;Y$;","
6100 PRINT"WHICH WHEN TRANSLATED MEANS:"
6110 PRINTY$;E$;" ";F$;Y$;". THE"
6120 PRINT"GREEKS WERE RIGHT. HAVING A"
6130 PRINT"SOUND MIND IN A SOUND"
6140 PRINTG$;" IS A PROPER GOAL FOR"
6150 PRINT"ALL ";H$;"-MINDED ";
6160 PRINT"PEOPLE."
6170 GOSUB11000:PRINT
6180 PRINT"TO ACHIEVE THIS, YOU MUST NOT"
6190 PRINT"ONLY BE IN ";I$;" SHAPE"
6200 PRINT"PHYSICALLY, BUT YOU MUST"
6210 PRINT"HAVE A";
6220 X$=J$:GOSUB8000:PRINTW$;
6240 PRINT" ";J$;" MENTAL"
6250 PRINT"ATTITUDE AND KEEP AN EYE ON YOUR";
6260 PRINTK$;" AT ALL TIMES."
6265 PRINT
6270 PRINT"IF YOU ARE TRULY INTERESTED IN"
6280 PRINT"SELF-IMPROVEMENT A GOOD MOTTO"
6290 PRINT"IS: ";Y$;L$;"!!!";Y$
6300 PRINT
6310 PRINT"
6320 GOSUB 12000
6330 IFPY$="R"THEN6000
7000 LPRINTTAB(7)"G O O D   H E A L T H   O R   E L S E !"
7020 LPRINT:LPRINT
7040 LPRINT"GOOD HEALTH IS MORE IMPORTANT ";
7050 LPRINT"THAN HAVING A LOT OF"
7060 LPRINTA$;" OR BEING A VERY ";
7070 LPRINT"IMPORTANT ";B$;". "
7080 LPRINT"THE OLD GREEKS USED TO ";
7090 LPRINT"SAY: ";Y$;C$;" ";D$;Y$;","
7100 LPRINT"WHICH WHEN TRANSLATED MEANS: ";
7110 LPRINTY$;E$;" ";F$;Y$;". THE"
7120 LPRINT"GREEKS WERE RIGHT. HAVING A ";
7130 LPRINT"SOUND MIND IN A SOUND"
7140 LPRINTG$;" IS A PROPER GOAL ";
7150 LPRINT"FOR ALL ";H$;"-MINDED"
7160 LPRINT"PEOPLE."
7170 LPRINT
7180 LPRINT"TO ACHIEVE THIS, YOU MUST NOT ";
7190 LPRINT"ONLY BE IN ";I$
7200 LPRINT"SHAPE PHYSICALLY, BUT YOU MUST ";
7210 LPRINT"HAVE A";
7220 X$=J$:GOSUB8000:LPRINTW$;
7240 LPRINT" ";J$;" ATTITUDE"
7250 LPRINT"AND KEEP AN EYE ON YOUR ";
7260 LPRINTK$;" AT ALL TIMES."
7270 LPRINT"IF YOU ARE TRULY INTERESTED IN ";

```



```

7280 LPRINT"SELF-IMPROVEMENT A GOOD MOTTO"
7290 LPRINT"IS: ";Y$;L$;"!!!";Y$
7300 LPRINT
7310 LPRINT"*****";
7320 LPRINT"*****"
7330 LPRINT:LPRINT
7340 GOSUB 12000
7350 IFPY$="R"THEN6000
7360 GOTO7000
8000 V$=LEFT$(X$,11)
8010 IFV$="A"ORV$="E"ORV$="I"ORV$="O"ORV$="U"THENW$="N"ELSEW$=""
8020 RETURN
11000 PRINT@482," ";
11020 KK$=INKEY$
11030 II$=INKEY$:IFII$<>" "THEN11030ELSECLS:RETURN
12000 PRINT@482," WE-READ PRINT-OUT MENU ";
12010 KK$ = INKEY$
12020 PY$ = INKEY$: IF PY$ = " " THEN 12020
12030 IFPY$="P"THENPRINT@482," ";
12040 IFPY$="M"THENRUN
12050 IFPY$="R"ORPY$="P"THENRETURNELSE12010
12060 RETURN

```



EXPLANANTION OF THE GRAPHICS CODES USED IN THE PROGRAM

Line	70	SHIFT + J's
Line	1000	CTRL + INV + TEXT: WRITE A STORY
Line	3000	CTRL + INV + TEXT: FUTURE PROFESSIONS
Line	3250	SPACES, then SHIFT + T's
Line	6020	CTRL + INV + TEXT: GOOD HEALTH OR ELSE
Line	6310	SPACES, then SHIFT + T's
Line	11000	CTRL + INV + TEXT: PRESS (SPACE) TO PROCEED
Line	12000	CTRL + INV + TEXT: first character of each word
Line	12030	CTRL + INV + TEXT: PRINTING STORY . . .

VZ-200 Hangman *

This program is the VZ-200 version of the traditional game. The computer chooses a word from its memory and prints on the screen a dash for each character of the selected word. You then choose which letter you want and if correct the computer will place that letter in the correct position(s). If you are wrong, then you begin the process of being hung! Another feature of the game is the ability to enter your own word instead of one from the computer.

FOR THE PROGRAMMER

If you have the 16K Memory expansion, you might like to increase the number of stored words. To do this simply add DATA lines and alter the value of NW in line 70 to match the total number of words stored.

Line 2035 checks to ensure that the player's finger has released the key before proceeding. The memory location 28416 should contain 255 when no key is pressed.


71-319
No words
more words

```
10 REM * VZ-200 HANGMAN *
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (APRIL 1983)
40 CLS
50 DIM A(26)
60 COLOR,0 '(SET BACKGROUND TO GREEN)
70 NW = 100 '(NUMBER OF WORDS)
320 DATA MOUTH,RIGHT,DEAD,WORM,CHEST,ROW, SORT,TRUE,CHEER,SON,SUN
330 DATA RUN,RAT,WORK,PIE,FENCE,PAPER,PEOPLE,MOTOR,BUSY,STIFF,IT
340 DATA PLEASE,DANCE,FRONT,MONTH,YEAR,DAY,HIGH,LOW,TODAY,WEATHER
350 DATA LADDER,BABY,CHALK,HEAVY,CLOSE,YOUNG,FIELD,EASY,TASTE,IN
360 DATA RADIO,FRIEND,PAST,WONDER,SUGAR,FOREST,KNOT
370 DATA LAMB,VOICE,TOUCH,KNIFE,RICE,GIANT,CLOTH,BRIDGE,FRESH,OF
380 DATA PIN,HEALTH,WEALTH,MISTAKE,NOTHING,COUNTRY,RED,BLUE
390 DATA OCEAN,SKIN,TOE,WORK,SPORT,BEAUTIFUL,UNDERSTAND,SOMEWHERE
400 DATA EVERYTHING,EVERYBODY,YESTERDAY,SOMETHING,BETWEEN,A
410 DATA ANYTHING,BREAKFAST,CHILDREN,ALMOST,SUMMER,AFTERNOON,SHUT
420 DATA CARPENTER,AMBULANCE,MEDICINE,MISERABLE,DAMAGE,ACCIDENT
430 DATA MACHINE,BATTERY,MONKEY,HORSE,PROTECTED,VERANDAH
460 REM SELECT WORD
470 FOR N = 1 TO RND(NW)
480 READ W$
490 NEXT
500 REM PRINT ALPHABET
510 COLOR 8
```

520 PRINT@65, "██████████"
 530 PRINT@97, "██████████"
 540 PRINT@129, "██████████"
 550 PRINT@161, "██████████"
 570 PRINT@47, "██████████"
 580 PRINT@108, "USE THE ARROW KEYS"
 590 PRINT@140, "TO MOVE THE CURSOR"
 600 PRINT@204, "USE THE <<CTRL>>"
 610 PRINT@236, "KEY TO SELECT LETTER"
 620 PRINT@289, "DO YOU WISH TO USE YOUR OWN"
 630 PRINT@ 321, "WORD (Y/N)? "
 635 K\$=INKEY\$
 640 I\$=INKEY\$:IFI\$=""THEN640
 650 IFI\$<>"Y"THEN CLS:GOTO 1000
 660 PRINT@289, "TYPE IN A WORD AND
 670 PRINT@321, "PRESS THE <<RETURN>> KEY
 680 PRINT@353, "(NO MORE THAN 14 LETTERS!)"
 690 PRINT:PRINT
 700 INPUT T\$
 710 IF T\$ <> "" THEN W\$ = T\$
 720 SOUND 1,1
 750 CLS



1000 REM PRINT ALPHABET
 1010 COLOR 8
 1020 PRINT@65, "██████████"
 1030 PRINT@97, "██████████"
 1040 PRINT@129, "██████████"
 1050 PRINT@161, "██████████"
 1060 PRINT@451, "ABCDEFGHIJKLMNOPQRSTUVWXYZ"
 1100 REM PRINT DASHES
 1110 COLOR 7
 1120 P = 387
 1130 FOR N = 1 TO LEN(W\$)
 1140 PRINT@P, "
 1150 P = P + 2
 1160 NEXT
 2000 REM SELECT LETTER
 2020 PO = 483
 2030 COLOR 8:PRINT@PO, "█";
 2035 IF PEEK(28416) <> 255 THEN 2035
 2040 PK = PEEK(28416)
 2060 IF PK = 223 THEN 3000
 2070 IF PK = 247 THEN 4000
 2080 IF PK = 251 THEN 5000
 2090 GOTO 2040
 3000 REM MOVE POINTER LEFT
 3010 PRINT@PO, " ";
 3020 PO = PO - 1
 3030 IF PO < 483 THEN PO = 483
 3040 GOTO 2030
 4000 REM MOVE POINTER RIGHT
 4010 PRINT@PO, " ";
 4020 PO = PO + 1



```


4030 IF PO > 508 THEN PO = 508
4040 GOTO 2030
5000 REM EXAMINE GUESS
5005 IF A(PO-482) = 99 THEN 7000
5010 L$ = CHR$(PO - 418)
5020 FOR L = 1 TO LEN(W$)
5030 IF MID$(W$,L,1) = L$ THEN GOSUB 6000
5040 NEXT
5050 IF C <> 99 THEN 10000 ' (DRAW BODY)
5060 C = 0 ' (RESET FLAG)
5070 IF LK = LEN(W$) THEN 20000 ' (WIN ROUTINE)
5080 GOTO 2030
6000 REM MOVE LETTER
6010 LP = 385 + (L*2) ' (LETTER POSITION)
6020 C = 99 ' (CORRECT FLAG)
6025 PRINT@PO-32,L$
6030 PRINT@PO,"";
6040 PRINT@PO-64,L$
6050 PRINT@PO-32,"";
6060 PRINT@PO," ";
6065 SOUND RND(20),2
6070 PRINT@PO,"";
6080 PRINT@PO-32," ";
6090 PRINT@PO,"";
6095 A(PO-482) = 99 ' (SET USED FLAG)
6100 IF PO - 96 > LP THEN 6200
6110 IF PO - 96 < LP THEN 6300
6120 IF PO - 96 = LP THEN 6400
6200 REM MOVE LETTER LEFT
6210 FOR N = PO - 64 TO LP + 32 STEP -1
6220 PRINT@N,L$
6230 PRINT@N+1," ";
6240 NEXT
6250 GOTO 6400
6300 REM MOVE LETTER RIGHT
6310 FOR N = PO - 64 TO LP + 32
6320 PRINT@N,L$
6330 PRINT@N-1," ";
6340 NEXT
6350 GOTO 6400
6400 REM MOVE LETTER UP
6402 IF C <> 99 THEN RETURN
6405 SOUND RND(10) + 10,2
6410 PRINT@LP,L$;
6420 PRINT@LP+32," ";
6430 LK = LK + 1 ' (LETTER COUNT)
6440 SOUND RND(10) + 20,2
6450 RETURN
7000 REM LETTER NOT THERE ROUTINE
7010 PRINT@PO,"";
7020 PRINT@PO-32,"";
7030 PRINT@PO," ";
7040 SOUND RND(20),2

```

```

7050 PRINT@PO,"";
7060 PRINT@PO-32," ";
7070 PRINT@PO,"";
7080 GOTO 2030
10000 REM INCORRECT GUESS
10010 IF RND(2) = 1 THEN LP = 384 ELSE LP = 415
10020 GOSUB 6030
10030 PRINT@416," ":PRINT@447," "
10040 SOUND 8,3:SOUND 4,3
10110 COLOR 8
11000 REM DRAW MAN
11010 M = M + 1
11020 IF M = 1 THEN 11100
11030 IF M = 2 THEN 11200
11040 IF M = 3 THEN 11300
11050 IF M = 4 THEN 11400
11060 IF M = 5 THEN 11600
11070 IF M = 6 THEN 11700
11080 IF M = 7 THEN 11800
11090 IF M = 8 THEN 11900
11095 IF M = 9 THEN 11950
11100 REM BASE
11105 COLOR 5
11110 PRINT@328,"";
11129 GOTO 2030
11200 REM UPRIGHT
11205 COLOR 5
11210 FOR N = 331 TO 10 STEP -32
11220 PRINT@N,"";
11230 NEXT
11240 GOTO 2030
11300 REM HORIZONTAL
11305 COLOR 6
11310 PRINT@12,"";
11320 GOTO 2030
11400 REM HEAD
11405 COLOR 2
11410 PRINT@54,"";
11420 FOR N = 86 TO 150 STEP 32
11430 PRINT@N,"";
11440 NEXT
11450 PRINT@182,"";
11460 PRINT@216,"";
11470 COLOR 3
11480 PRINT@97,"";
11490 COLOR 7
11500 PRINT@120,"";
11510 COLOR 4
11520 PRINT@151,"";
11530 GOTO 2030
11600 REM TRUNK
11610 COLOR 2

```




```

12310 K$ = INKEY$
12320 I$ = INKEY$:IF I$ = "" THEN 12320
12330 IF ASC(I$) <> 13 THEN 12320 ELSE RUN
15000 REM WRIGGLE BODY
15010 GOTO 2030
15020 RETURN
20000 REM WIN ROUTINE
20100 PRINT@112,"PHEW!"
20110 FOR N = 1 TO 4
20120   SOUND 23,1
20130 NEXT
20140 FOR N = 1 TO 4
20150   SOUND 25,1
20160 NEXT
20170 SOUND 23,1
20180 SOUND 23,1
20190 SOUND 27,1
20200 SOUND 27,1
20210 SOUND 28,4
20220 GOTO 12260

```

EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

Line	520	SHIFT + U, then SHIFT + T's, then SHIFT + I
Line	530	SHIFT + U, then CTRL + INV + TEXT: VZ-200, then SHIFT + I
Line	540	SHIFT + U, then CTRL + INV + TEXT: HANGMAN, then SHIFT + I
Line	550	SHIFT + U, then SHIFT + Y's, then SHIFT + I
Line	570	CTRL + INV + TEXT: INSTRUCTIONS
Line	1020	SHIFT + U, then SHIFT + T's, then SHIFT + I
Line	1030	SHIFT + U, then CTRL + INV + TEXT: VZ-200, then SHIFT + I
Line	1040	SHIFT + U, then CTRL + INV + TEXT: HANGMAN, then, SHIFT + I
Line	1050	SHIFT + U, then SHIFT + Y's, then SHIFT + I
Line	2030	SHIFT + J
Line	6030	SHIFT + T
Line	6050	SHIFT + J
Line	6070	SHIFT + T
Line	6090	SHIFT + J
Line	7010	SHIFT + T
Line	7020	SHIFT + J
Line	7050	SHIFT + T
Line	7070	SHIFT + J
Line	11110	SHIFT + Y's
Line	11220	SHIFT + J

Banton
He did this i's Hack

code explanation cont.

Line 11310 SHIFT + T's, then SHIFT + J
Line 11410 SHIFT + U, then SHIFT + T's, then SHIFT + I
Line 11430 SHIFT + U, then SPACES, then SHIFT + I
Line 11450 SHIFT + U, the SHIFT + T's, then SHIFT + I
Line 11460 SHIFT + T
Line 11480 SHIFT + Y, then a SPACE, then SHIFT + Y
Line 11500 SHIFT + Y
Line 11520 SHIFT + Y's
Line 11620 SHIFT + Y's, then SHIFT + J, then SHIFT + Y's
Line 11630 SHIFT + J's
Line 11640 SHIFT + J's
Line 11720 SHIFT + G
Line 11730 SHIFT + G
Line 11740 SHIFT + J
Line 11820 SHIFT + H
Line 11830 SHIFT + H
Line 11840 SHIFT + J
Line 11920 SHIFT + H
Line 11930 SHIFT + H
Line 11935 SHIFT + T's
Line 11970 SHIFT + G
Line 11980 SHIFT + G
Line 11990 SHIFT + T's
Line 12300 CTRL + INV + TEXT: PRESS (RETURN) FOR NEW GAME



Moving Targets

In this program you have to try to hit the moving targets. Each one hit scores 500 points, while each miss will cost you 100 points. A score of over 8,000 will give you bonus time. As the score gets higher, you will find that the targets become harder to hit.


The computer will keep track of the name and score of the player with the highest score.

FOR THE PROGRAMMER

This program has been designed in a straightforward manner to allow you to follow the logic and see how the 'target' is moved in lines 5010-5040. Also the determination as to whether the target has been hit or not (line 6400- 6440) and progressive scoring in line 2010.

The logic used in this program can be 'tailored' to suit a variety of applications that require movement of an object, so examine the routines closely.

```
10 REM * MOVING TARGETS *
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (MAY 1983)
35 COLOR ,0
40 CLS
50 D = 2000
60 GOSUB 30000 '(INTRODUCTION)
70 SOUND 23,1:SOUND 23,1
80 SOUND 28,3
90 CLS
400 PRINT@1,"SCORE ="
410 PRINT@18,"TIME   ="
420 PRINT@50,"HITS   ="
430 PRINT@82,"MISSES ="
440 S = 0:H = 0:M = 0
450 IF TS = 0 THEN 500
460 PRINT@33,"CHAMP ="
470 PRINT@41,;:PRINTUSING"##,###";TS
500 T$(1) = "■":T$(2) = "■"
520 DF = 8
540 COLOR 8
550 LP = 496
560 PRINT@LP,"■";
570 GOSUB 600
```



```

580 TI = 1000 '(START COUNTDOWN AT 1000)
590 GOTO 1000
600 REM RESET STARTING POINTS
610 T = 1:D = 100
620 TP = RND(32) + 126
630 C = RND(6) + 2
640 IF S > 3000 THEN 650 ELSE RETURN
650 IF S > 5000 THEN 660 ELSE DF = 6:RETURN
660 IF S > 7000 THEN 670 ELSE DF = 5:RETURN
670 IF S > 9000 THEN 680 ELSE DF = 4:RETURN
680 IF S > 11000 THEN 690 ELSE DF = 3:RETURN
690 IF S > 12000 THEN 700 ELSE DF = 2:RETURN
700 DF = 1:RETURN
1000 GOSUB10000
1010 GOSUB 2000
1020 GOTO 5000
2000 REM UPDATE SCORES
2010 TI = TI - 2
2020 PRINT@27,,:PRINTUSING"####";TI
2030 IF TI = 0 THEN 20000
2040 RETURN
5000 REM MOVE LASER
5010 PK = PEEK(28416)
5020 IF PK = 223 THEN 5100 '(MOVE LEFT)
5030 IF PK = 247 THEN 5300 '(MOVE RIGHT)
5040 IF PK = 251 THEN 6000 '(FIRE)
5050 GOTO 1000
5100 LP = LP - 1
5110 IF LP < 481 THEN LP = 481
5120 COLOR 8
5130 PRINT@LP,"█" ;
5140 PRINT@LP + 1," " ;
5150 GOTO 1000
5300 LP = LP + 1
5310 IF LP > 510 THEN LP = 510
5320 COLOR 8
5330 PRINT@LP,"█" ;
5340 PRINT@LP - 1," " ;
5350 GOTO 1000
6000 REM FIRE CANNON
6005 COLOR 3:PRINT@LP,"██";
6010 COLOR 4:SOUND 15,1:GOSUB 6400
6020 FOR N = LP - 160 TO LP - 352 STEP - 96
6030 PRINT@N,"███"
6040 PRINT@N+96," "
6050 NEXT
6055 COLOR 3:PRINT@LP,"██";
6060 COLOR 2
6070 PRINT@N+96," "
6100 IF HT <> 99 THEN 6200 ELSE HT = 0
6110 COLOR 8
6120 PRINT@TP,"███"
6130 SOUND 30,1:SOUND 29,1
6140 PRINT@TP," "

```



```

21040 CLS
21050 SOUND 23,1:SOUND 23,1
21060 SOUND 28,3
21070 GOTO 60
30000 REM DRAW FRAME
30005 CLS:COLOR 8
30010 PRINT@0,"
30020 PRINT@480,"
30030 FOR N = 32 TO 448 STEP 32
30040 PRINT@N," ";
30050 PRINT@N+31," ";
30060 NEXT
30070 PRINT@31," ";
30080 POKE 29183,248
30100 COLOR 7
30110 PRINT@72,"
30120 T$ = "(BY WAYNE G. RICHMOND)"
30130 FOR N = 1 TO LEN(T$)
30140 PRINT@133,RIGHT$(T$,N);
30150 NEXT
30160 PRINT@194,"* AIM WITH THE ARROW KEYS";
30170 PRINT@226,"* FIRE WITH THE <<CTRL>> KEY";
30180 PRINT@290,"* OVER 8,000 GETS BONUS TIME";
30182 PRINT@322,"* MISSES COST 100 POINTS!";
30185 IF TS = 0 THEN 30220
30190 PRINT@354,"* CURRENT CHAMPION IS:";
30200 PRINT@418,T$, " ";PRINTUSING" ##,###";TS
30210 GOTO30400
30220 PRINT@386,"* TARGETS BECOME HARDER TO";
30230 PRINT@418," HIT AS YOUR SCORE RISES!";
30400 FOR N = 1 TO D:NEXT
30410 K$ = INKEY$
30420 PRINT@131," <<HIT ANY KEY TO BEGIN>>";
30430 SOUND 30,1
30440 FOR N = 1 TO 80
30450 I$=INKEY$:IFI$<>" THEN RETURN
30460 NEXT
30470 PRINT@132,"
";
30480 FOR N = 1 TO 20
30490 I$=INKEY$:IFI$<>" THEN RETURN
30500 NEXT
30600 GOTO 30420

```

For explanation of graphics codes used, see next page.

EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

Line 500 SHIFT + A, SHIFT + T and SHIFT + S. SHIFT + J
Line 560 SHIFT + J
Line 5130 SHIFT + J
Line 5330 SHIFT + J
Line 6005 SHIFT + Y
Line 6030 SHIFT + T
Line 6055 SHIFT + J
Line 6120 SHIFT + D and SHIFT + Y and SHIFT + F
Line 20040 CTRL + INV + TEXT: BONUS
Line 20100 CTRL + INV + TEXT: GAME OVER
Line 21000 SHIFT + U, then CTRL + INV + TEXT: PRESS (SPACE)
FOR NEW GAME, then SHIFT + I
Line 30010 SHIFT + Y's
Line 30020 SHIFT + T's
Line 30040 SHIFT + I
Line 30050 SHIFT + U
Line 30070 SHIFT + S
Line 30110 SHIFT + U, then CTRL + INV + TEXT: MOVING TARGETS
then SHIFT + I
Line 30200 CTRL + INV + TEXT: SCORE:



VZ-200 Mastermind



This is the computerised version of one of the world's most popular games, Mastermind. In this version, you are pitted against the amazing VZ-200 . . . Can you win?

The standard rules for Mastermind apply here. You have to choose from eight colours depicted along the top row of the keyboard. Simply touch the key below the required colour and it will be selected.

For each colour correctly selected and placed in the correct position, you will receive a black marker. A correct colour only (and in the wrong position) will result in a white marker being displayed. These are the only clues you will be given. You will have only eight turns to try to deduce the correct combination. Good luck!

N.B. If you have trouble differentiating between the colours on your TV screen, you may wish to replace line 1530 with the REM statement in line 40 (remember to remove the REM word itself). This will display the colour number next to each colour square.

FOR THE PROGRAMMER

As this program is a little more complex than most, we have again used REM statements throughout the program to explain what is happening in the various sections. Note the way in which the initial random code is generated in lines 200-280. Also, the continued use of string functions throughout this program.

Also note the 'guess' routine from lines 1000-1100 which incorporates the keyboard Peek address. It (the PEEK) is 'calling' the keyboard to check and 'see' whether or not a key has been pressed and if so, which key.

```
10 REM VZ-200 MASTERMIND
20 REM WRITTEN BY WAYNE G. RICHMOND
30 REM (APRIL, 1983)
40 REM ALTERNATIVE LINE 1530 PRINTEV+H, I$; "####";
50 CLS
100 V = -24 '(SET KNOB VERTICAL)
110 VP = -8 '(SET PEG VERTICAL)
120 COLOR ,0
200 REM SELECT RANDOM CODE
210 FOR N = 1 TO 4
220   X = RND(8)
230   X$ = MID$(STR$(X),2,1)
240   FOR A = 1 TO LEN(R$)
250     IF X$ = MID$(R$,A,1) THEN 220
260   NEXT
```



```

270 R$ = R$ + X$
280 NEXT
300 REM INTRODUCTION
305 COLOR 8
308 PRINT@7, " "
310 PRINT@39, " "
315 PRINT@71, " "
318 SOUND 17,1
320 T$="(BY WAYNE G. RICHMOND)"
330 FOR N = 1 TO LEN(T$)
340 PRINT@101,RIGHT$(T$,N);
345 FOR X = 1 TO 5:NEXT
350 NEXT
355 SOUND 18,1
360 PRINT@161,"CAN YOU DEDUCE THE 4 COLOUR"
370 PRINT@193,"CODE IN 8 TRIES?"
380 PRINT
390 PRINT" USE THE 8 NUMBER/COLOUR KEYS"
400 PRINT
410 COLOR 3
420 PRINT" # = CORRECT COLOUR & POSITION"
430 PRINT:COLOR 5
440 PRINT" # = CORRECT COLOUR ONLY"
450 K$ = INKEY$
460 COLOR 8
480 PRINT@453, " "
490 SOUND 30,1
500 FOR N = 1 TO 80
510 I$ = INKEY$:IF I$ <> "" THEN CLS:GOTO 1000
520 NEXT
530 PRINT@453, " "
540 FOR N = 1 TO 20
550 I$ = INKEY$:IF I$ <> "" THEN CLS:GOTO 1000
560 NEXT
570 GOTO480
1000 REM INPUT GUESS
1010 H=0:V=V+64:HP=0:VP=VP+64
1020 C$="":B=0:W=0
1025 IF PEEK(28416) <> 255 THEN 1025
1030 P1 = PEEK(28663):I = 0
1040 IF P1 = 255 THEN 1200
1050 IF P1 = 239 THEN I = 1
1060 IF P1 = 253 THEN I = 2
1070 IF P1 = 247 THEN I = 3
1080 IF P1 = 223 THEN I = 4
1090 IF P1 = 254 THEN I = 5
1100 IF I = 0 THEN 1030
1110 GOTO 1500
1200 P2 = PEEK(28639)
1210 IF P2 = 255 THEN 1030
1220 IF P2 = 254 THEN I = 6
1230 IF P2 = 223 THEN I = 7
1240 IF P2 = 247 THEN I = 8

```

```

1250 IF I = 0 THEN 1030
1500 I$ = MID$(STR$(I),2,1):GOTO 1800
1510 C$ = C$ + I$
1520 COLOR I
1530 PRINT@V+H," ■■■";
1535 SOUND 30,1
1540 IF H = 12 THEN 2000
1550 H = H + 4
1560 IF PEEK(28416) <> 255 THEN 1560
1570 GOTO 1030
1800 FOR N = 1 TO LEN(C$)
1810 IF I$ = MID$(C$,N,1) THEN F = 99
1820 NEXT
1830 IF F = 99 THEN F = 0:GOTO 1030
1840 GOTO 1510
2000 REM EXAMINE ANSWER
2010 FOR R = 1 TO 4
2020 FOR C = 1 TO 4
2030 IF MID$(R$,R,1) = MID$(C$,C,1) THEN GOSUB 3000
2040 NEXT C
2050 NEXT R
2060 IF B = 0 AND W = 0 THEN SOUND 7,3:SOUND 4,3
2100 REM DRAW PEGS
2110 IF B = 0 THEN 2200
2120 COLOR 3
2130 FOR N = 1 TO B
2140 PRINT@V+HP," ■ ";
2145 SOUND 20,1
2150 HP=HP+2
2160 NEXT
2200 IF W = 0 THEN 2300
2210 COLOR 5
2220 FOR N = 1 TO W
2230 PRINT@V+HP," ■ ";
2235 SOUND 10,1
2240 HP=HP+2
2250 NEXT
2300 IF B = 4 THEN 12000 '(WINNER)
2310 IF V = 488 THEN 12000 '(LOSER)
2320 IF PEEK(28416) <> 255 THEN 2320
2330 GOTO 1000
3000 IF R = C THEN B=B+1 ELSE W=W+1
3010 RETURN
8000 PRINT PEEK(28416):GOTO 8000
12000 REM PRINT CORRECT CODE
12010 PO=33
12020 FOR N = 1 TO 4
12025 M$ = MID$(R$,N,1)
12030 K = VAL(M$)
12040 COLOR K
12050 PRINT@PO," ■ ";
12060 PRINT@PO+32," ■ ";
12065 PRINT@PO-32,M$;

```



```

12070 PO = PO + 2
12080 NEXT
12100 IF B <> 4 THEN SOUND 1,9:GOTO 12200
12110 SOUND 16,3
12120 SOUND 11,2
12130 SOUND 11,1
12140 SOUND 13,3
12150 SOUND 11,3
12160 SOUND 0,2
12170 SOUND 15,4
12180 SOUND 16,4
12200 PRINT@161, "PRESS";
12210 PRINT@225, "RETURN";
12220 PRINT@289, "FOR";
12230 PRINT@353, "NEW";
12240 PRINT@417, "GAME";
12245 K$ = INKEY$
12250 I$=INKEY$:IFI$=""THEN12250
12260 IF ASC(I$) <> 13 THEN 12250
12270 SOUND 13,1
12280 RUN

```

EXPLANATION OF THE GRAPHICS CODES USED IN THE PROGRAM

Line	40	SHIFT + J's
Line	308	SHIFT + Y's
Line	310	CTRL + INV + TEXT: VZ-200 MASTERMIND
Line	315	SHIFT + T's
Line	420	SPACE, then SHIFT + U
Line	440	SPACE, then SHIFT + U
Line	480	CTRL + INV + TEXT: HIT ANY KEY TO BEGIN
Line	1530	SHIFT + J's
Line	2140	SHIFT + U
Line	2230	SHIFT + U
Line	12050	SHIFT + J
Line	12065	SHIFT + J
Line	12200	CTRL + INV + TEXT: PRESS
Line	12210	CTRL + INV + TEXT: RETURN
Line	12220	CTRL + INV + TEXT: FOR
Line	12230	CTRL + INV + TEXT: NEW
Line	12240	CTRL + INV + TEXT: GAME

VZ-200 Cup *


This neat little program simulates a horse race. Up to ten people can place bets on any of the six horses. Each horse is given 'odds' and as a rule will race accordingly. However a long shot has been known to win!

Imagine the joy of tearing your hair out as you watch the race in living colour. The computer will keep track of the bettors' names and winnings (if any!) and is one program that should keep you on the edge of your chair.

FOR THE PROGRAMMER

Lines 4000-4060 determine whether or not a particular horse will move on each round. Line 4010 lessens the chance of a horse with high odds while line 4007 gives that same horse an extra chance. Line 575 however, is one that can on occasions give a 'long-shot' horse a real winning chance.

```
10 REM * VZ-200 CUP *
20 REM BY WAYNE G. RICHMOND (MAY 1983)
30 REM (MAY 1983)
50 CLS:COLOR ,0:CLEAR 300
80 DIM N$(6), C(6), H$(6), P(6), M(24), P$(6)
90 SP$=" " " ":GOTO 500
200 K$ = INKEY$
210 I$ = INKEY$:IF I$ <> " " THEN 210 ELSE RETURN
500 REM SET-UP
510 FOR N = 1 TO 6
520   READ N$(N) '(NAME)
530   READ C(N) '(COLOUR CODE)
540   READ H$(N) '(HORSE)
550   READ P(N) '(STARTING POS)
560   D(N) = RND(10) '(ODDS)
570   DD(N) = D(N) '(MOVEMENT ODDS)
575   IF DD(N) < 3 AND RND(3) = 1 THEN DD(N) = 10' (LONG SHOT)
580   R = R + D(N)
590 NEXT
600 FOR N = 1 TO 6
610   READ P$(N)
620 NEXT
660 GOSUB 30000 '(INTRO)
700 GOSUB 8000 '(BETS)
1000 REM SET-UP TRACK
1010 GOSUB 12000 '(CLEAR SCREEN)
1040 COLOR 1
1050 FOR N = 93 TO 413 STEP 32
1060 PRINTON, " ";
1070 NEXT
```



```

1100 FOR N = 1 TO 6
1110   COLOR C(N)
1120   PRINT@P(N),H$(N);
1140 NEXT
1200 PRINT@450,"THEY'RE AT THE GATES . . .";
1210 SOUND 11,6:SOUND 11,3:SOUND 11,3:SOUND 11,3:SOUND 16,6
1300 PRINT@450,"THEY'RE OFF!! . . . ";
2000 REM RACE ROUTINE
2010 FOR N = 1 TO 6
2020   GOSUB 4000
2030   COLOR C(N)
2040   PRINT@P(N)+M(N),H$(N);
2060 NEXT
2070 IF W = 99 THEN 7000' (CHECK FOR WINNER(S))
2100 GOTO 2010
4000 REM DECIDE ON MOVEMENT
4005 IF RND(3) = 1 THEN RETURN
4007 IF RND(4) = 1 THEN IF DD(N) < 2 THEN 4020
4010 IF RND(DD(N)) < 2 THEN 4030
4020 M(N) = M(N) + 1:GOTO 4040
4030 IF RND(3) = 1 THEN M(N) = M(N) + 1
4040 IF M(N) = 24 THEN W = 99
4050 IF M(N) > 3 THEN PRINT@450,"
4060 RETURN
7000 REM WINNER
7010 FORN=1TO4:SOUND28,1:SOUND25,1:NEXT:SOUND28,6
7020 PRINT@440,"
7030 GOSUB 200:GOSUB 12000:COLOR 5
7090 PRINT@65,"
7100 P=1:PO=130:WN=0
7110 FOR X = 24 TO 1 STEP -1
7120   FOR N = 1 TO 6
7130     IF M(N) <> X THEN 7190
7140     WN = 99
7150     COLOR C(N)
7160     PRINT@PO,P$(P);" ";H$(N);" ";N$(N);
7170     PRINT@PO+23,;:PRINTUSING"# #:1";INT(R/D(N))
7180     PO = PO + 32
7190   NEXT N
7200   IF WN = 99 THEN P = P + 1:WN = 0
7210 NEXT X
7220 COLOR 7
7230 PRINT@417,"
7240 GOSUB 200:GOSUB 9000:COLOR 7
7270 PRINT@417,"
7280 IS = INKEY$:IF IS = " " THEN 7280
7290 IF IS = "N" THEN RUN ELSE IF IS <> "Y" THEN 7280
7300 REM SET-UP FOR NEW RACE
7310 GOSUB 12000
7320 R=0:W=0
7330 FOR N = 1 TO 6
7340   D(N) = RND(10)
7350   DD(N) = D(N)

```

```

7360 IF DD(N) < 3 AND RND(3) = 1 THEN DD(N) = 10
7370 R = R + D(N)
7380 M(N) = 0
7390 NEXT
7400 SG = 99 '(SECOND GAME FLAG)
7410 GOTO 700
8000 REM SET-UP TRACK
8010 GOSUB 12000:COLOR 5
8050 PRINT@57, " " ;
8060 PO = 98
8100 FOR N = 1 TO 6
8105 PO = PO + 32
8110 COLOR C(N)
8120 PRINT@PO, H$(N); " "; N$(N);
8130 PRINT@PO+19, ;:PRINTUSING"##:1";INT(R/D(N))
8140 NEXT
8145 IF SG = 99 THEN 8260
8150 PRINT@355, "HOW MANY BETTORS (1-10)";
8160 INPUT NB$:IF NB$ = "" THEN 8150
8170 NB = VAL(NB$)
8180 IF NB < 1 OR NB > 10 THEN PRINT@379, " ";:GOTO 8150
8190 PRINT@355, "ENTER NAMES OF BETTORS ";
8200 FOR N = 1 TO NB
8210 PRINT@419, SP$;
8220 PRINT@419, "BETTOR #";N;
8230 INPUT BN$(N)
8240 IF BN$(N) = "" THEN 8220
8245 BN$(N) = LEFT$(BN$(N),14)
8250 NEXT
8260 PRINT@355, "ENTER HORSE NUMBER AND BET ";
8270 FOR N = 1 TO NB
8275 PRINT@387, "FOR ";BN$(N);
8280 PRINT@419, SP$;
8300 PRINT@419, "HORSE #";
8310 INPUT HN$
8320 IF HN$ = "" THEN 8300
8330 HN(N) = VAL(HN$)
8335 IF HN(N) < 1 OR HN(N) > 6 THEN 8280
8340 PRINT@431, "BET ($)";
8350 INPUT A$
8360 IF A$ = "" THEN 8340
8370 BA(N) = VAL(A$)
8380 IF BA(N) < 1 THEN 8340
8390 NEXT
8400 RETURN
8500 GOTO8500
9000 REM WINNINGS
9010 FOR N = 1 TO NB
9020 IF M(HN(N))=24 THEN W(N)=W(N)+INT((R/D(HN(N))))*BA(N)
9030 IF M(HN(N)) <> 24 THEN W(N) = W(N) - BA(N)
9040 NEXT
9100 GOSUB 12000
9110 PRINT@66, " " ;

```

8100

14
90

```

9120 PO = 130
9130 FOR N = 1 TO NB
9140   PRINT@PO,;
9160   PRINTUSING"%           %+$\$##### ,###";BN$(N);W(N)
9170   PO = PO + 32
9180 NEXT
9190 RETURN
9999 GOTO9999
12000 REM CLEAR SCREEN
12010 FOR N = 65 TO 449 STEP 32
12020   PRINT@N,SP$;" ";
12030 NEXT
12040 RETURN
30000 REM INTRODUCTION
30010 GOSUB 31000' (DRAW FRAME)
30090 COLOR 7
30100 PRINT@74,"";
30110 T$ = "(BY WAYNE G. RICHMOND)"
30120 FOR N = 1 TO LEN(T$)
30130   PRINT@133,RIGHT$(T$,N);
30140 NEXT
30150 PRINT@194,"THE BIG RACE IS ABOUT TO";
30160 PRINT@226,"START!!  CHOOSE YOUR HORSE";
30170 PRINT@258,"AND LAY YOUR BETS AND WATCH";
30180 PRINT@290,"THEM PONIES GO!!";
30190 PRINT@418,"";
30200 GOSUB 200
30210 SOUND 30,1
30215 RETURN
31000 REM DRAW FRAME
31005 COLOR 8
31010 PRINT@0,"";
31020 PRINT@480,"";
31030 FOR N = 32 TO 448 STEP 32
31040   PRINT@N," ";
31050   PRINT@N+31," ";
31060 NEXT
31070 PRINT@31," ";
31080 POKE 29183,248
31090 RETURN
40000 DATA "JIMMY BOY",4,"",65
40010 DATA "NOSY PARKER",3,"",129
40020 DATA "FRISKY JOE",8,"",193
40030 DATA "SNOW JOKE",5,"",257
40040 DATA "TRICKY DICKY",2,"",321
40050 DATA "WHATA DOLL",7,"",385
40060 DATA "1ST","2ND","3RD","4TH","5TH","6TH"

```



CM →



For explanation of graphics codes used, see next page.